

**SANYO**

パーソナルコンピュータ 拡張BASIC説明書 **MSX2+**   
WAVYシリーズ

**WAVY**



# はじめに

この拡張BASIC説明書は、MSXおよびMSX2をさらに機能アップしたMSX2+およびMSX-MUSIC、BASICコンパイラのBASIC命令の説明書です。

MSX2+またはMSX-MUSIC、BASICコンパイラのどの機能が装備されているかは、お買い求めのパソコンによって異なります。別冊の取扱説明書をお読みいただき、各機能があるかを確認してください。

この説明書をご使用になる前に、つなぎ方、基本的な操作方法について、別冊の取扱説明書をよくお読みください。

また、BASICコンパイラを使う前にBASIC命令の使い方とプログラムの作り方を覚えておいてください。プログラムの作り方については別冊の「MSXプログラミング入門」をお読みください。

- ご注意
- (1)本書の内容の一部又は全部を無断で転載することは禁止されています。
  - (2)本書の内容については、将来予告なしに変更することがあります。
  - (3)本書の内容を運用した結果の影響については、一切の責任を負いかねますのでご了承ください。

MSXは、アスキーの登録商標です。

# 目次

この拡張 BASIC マニュアルは、次の 4 章に分けて使い方を説明しています。目的に合わせてお読みください。

## MSX<sub>2+</sub> の使い方 ..... 3

MSX<sub>2+</sub> てなに? -こんなことができます- ..... 4

漢字の表示 ..... 6

漢字入力のしかた、記号入力のしかた、こんなこともできます、プログラム中の漢字モード、プログラムを作るときの注意

多彩なカラー表示 ..... 13

スクリーンの種類、色の再現方法 -YJK方式とは-、YJK方式のしくみ

MSX<sub>2+</sub> のコマンド -MSX、MSX<sub>2</sub>との違い- ..... 20

機能別索引、コマンド解説

## MSX-MUSIC の使い方 ..... 39

FM音源とは? ..... 40

音楽命令の使い方 -コマンド一覧 ..... 42

機能別索引、コマンド解説

## BASIC コンパイラの使い方 ..... 59

コンパイラとは? -コンパイラの特長- ..... 60

コンパイラの使い方 ..... 62

コンパイラの使い方、コンパイラ用の命令、使える命令と使えない命令、こんなこともできます、プログラム作成時のご注意

## 資料 ..... 75

サンプルプログラム ..... 76

索引 ..... 80

MSX  
てなに?  
漢字の表示  
多彩な  
カラー表示  
MSX<sub>2</sub>  
コマンドの  
FM音源  
とは?  
音楽命令  
の使い方  
コンパイラ  
とは?  
コンパイラ  
の使い方  
資料





# MSX<sup>2+</sup>の使い方



この章では、MSXおよびMSX2に対して追加・変更されたMSX BASIC Ver. 3の拡張機能について説明します。

漢字入力、多彩なカラー表示など、MSX2+パソコンの機能をフルに使うためにBASIC命令の使い方を覚えてください。

MSX2+ ってなに? -こんなことができます-	4
漢字の表示	6
● 漢字入力のしかた	6
● 記号入力のしかた	8
● こんなこともできます	9
● プログラム中の漢字モード	10
● プログラムを作るときの注意	11
多彩なカラー表示	13
● スクリーンの種類	13
● 色の再現方法 -YJK方式とは-	14
● YJK方式のしくみ	16
MSX2+のコマンド -MSX,MSX2との違い-	20
● 機能別索引	20
● コマンド解説	22

# MSX<sub>2</sub><sup>プラス</sup>ってなに？ -こんなことができます-

別冊の「MSXプログラミング入門」ではMSX BASICの命令の基本的な使い方を説明しています。この章ではMSX、MSX2に追加・変更されたMSX2+の機能を説明します。

MSX2+は、MSX2に漢字の入出力機能や19,268色のカラー表示機能などを追加した新しいBASICです。

主に次の機能が追加されています。

- 漢字の入出力ができます。

JIS第1水準の漢字ROMを標準で備えています。単漢字変換による漢字混じりのプログラム入力や、PRINT命令による漢字の表示が簡単にできます。

- 多彩なカラー表示ができます。


スクリーンモードの追加により、最高19,268色の表示ができます。自然画の表示もよりリアルに行えます。

- 表示画面を自由にスクロールできます。

画面のスクロール(移動)が、縦方向に加えて横方向にもできます。ゲームなどで縦、横、斜め自由に移動できます。

このほかにも、漢字対応プリンタに、漢字混りのプログラムリストを出したりすることもできます。

また、MSX2+のオプション機能(標準ではなく、パソコンごとに追加できる機能)として、JIS第2水準の漢字ROMへの対応や、MSX-MUSICのBASIC命令にも対応できます。





## こんなことができます


### 漢字の入出力

- 漢字混りのプログラム

```
Ok  
list  
10 SCREEN 2  
20 CALL KANJI  
25 LOCATE 1,5  
30 PRINT "MSX2+は漢字が出ます"  
40 LINE (0,0)-(255,191),15  
50 GOTO 50  
Ok  
■  
  
color auto goto list run
```


漢字のプログラム入力、プリンタへの出力ができます。

- 漢字とグラフィックの混在



漢字をグラフィック画面に重ねて表示できます。

### 19,268色の表示



自然画をリアルに再現できます。

### 画面の水平スクロール



縦、横、斜めに画面を移動できます。

# 漢字の表示

MSX2+パソコンでは、JIS第一水準の漢字ROMを標準で備えています。テキスト画面で単漢字変換(1つの漢字の音読みを入力して漢字にする)を使った漢字の入力ができ、漢字を使ったプログラムの作成ができます。また、漢字対応のプリンタに漢字出力ができます。(第二水準の漢字ROMは、標準装備ではありません。MSX2+のオプション装備です。機器の仕様を確かめてください)。

## 漢字入力のしかた —「漢字」と入力するとき—

- 1 <sup>コール</sup>CALL <sup>カンジ</sup>KANJI ↵(リターンキー)と入力します。(漢字モード)

画面がクリアされ、「OK」という文字が漢字モードで大きく表示されます。

「CALL」の代わりに「\_ (アンダーバー)」でもできます。

(\_KANJI ↵)

CALLKANJI ↵



- 2 <sup>コントロール</sup>CTRLキーを押しながらスペースキーを押します。(単漢字変換モードになります)

画面の下に変換テーブルと呼ぶ白色の帯が表示されます。(文字の表示色(COLORの色)を変えるとこのテーブルの色も変わります)。

OK

color auto goto list run



- 3 **かな**キーを押して、かな入力の状態にします。(ローマ字カナ入力もできます)。

OK



- 4 次に、「か」と入力します。

画面の下にひらがなの「か」と、「か」で始まる漢字が表示されます。音読みが「か」の漢字が反転して表示されます。

OK

か **化**下家可価歌加科火



- 5 「ん」と入力します。


画面の下にひらがなの「かん」と、「かん」で始まる漢字が表示されます。音読みが「かん」の漢字が反転して表示されます。

OK


かん **間**関完館官観管感




**6** 変換テーブル内のカーソルをカーソルキーを押して、目的の文字に移動します。もし、目的の漢字がないときは、カーソルキー↑または↓を押すと別の変換テーブルが表示されます。(次ページの「ちょっと一言」を参照)




**7** ⏪(リターンキー)を押すと、「漢」が画面に表示されます。(変換テーブル内の文字はクリアされます)。



**8** 同様にして手順**4**~**7**の要領で「じ」を入力して、漢字の「字」を表示させます。



**9** **CTRL**キーを押しながらスペースキーを押します。変換テーブルが消え、漢字入力を終了します。



**ちょっと一言**

■ 単漢字変換とは

ワープロやパソコンでは直接漢字を入力できないので、漢字の読み仮名を入力して漢字に置き換え(変換)します。単漢字変換とは、漢字を単一(一文字ずつ)入力して、置き換える方法のことです。

■ 漢字を表示させる命令

漢字を表示させる命令として、次の命令があります(26ページを参照)。

BASIC 命令	CALL KANJI CALL KANJI0	CALL KANJI1	CALL KANJI2	CALL KANJI3
ドット数	16X16ドット	12X16ドット	16X16ドット	12X16ドット
画面表示	1画面表示 (ノンインターレス)	1画面表示 (ノンインターレス)	2画面交互表示 (インターレス)	2画面交互表示 (インターレス)

## 記号入力のかた - ☆を入力するとき -

"●"や"★"などの記号には読み方がないので、漢字コードで指定します。漢字コードは別冊の「MSXプログラミング入門」の「5. 資料」にある、JIS漢字コード表の番号(4ケタ)です。

- 1 CALL KANJI ↵(リターンキー)と入力します。
- 2 **CTRL**キーを押しながらスペースキーを押します。(単漢字変換モードになります)
- 3 **CTRL**キーを押しながら**2179**と押します。(漢字コードの入力)  
画面の下に図のように記号が表示されます。
- 4 変換テーブル内のカーソルをカーソルキーで移動させ、目的の記号に合わせます。
- 5 ↵(リターンキー)を押すと、記号が画面に表示されます。(変換テーブル内の記号はクリアされます)
- 6 **CTRL**キーを押しながらスペースキーを押します。

CALL KANJI ↵



**CTRL** + **スペース**

**CTRL** + **2179**

**2179**から表示される。

☆ ★ ○ ● ◎ ◆ □ ■

### ちょっと ■ 変換テーブルに表示される文字数

漢字の読みを入力したときや記号の漢字コードを入力したとき、変換テーブルに漢字が表示されます。一度に表示される文字数は、画面の表示文字数で決まり、CALL KANJI3、SCREEN 0でWIDTH 80のときが最大で、15文字です。

また、漢字コードを入力したときなど、変換テーブルに表示しきれない文字はバッファ(数値やデータを一次的に記憶しているところ)に貯えられ、カーソルキーを押してカーソルを移動させると順に表示されます。一つの読み(コード)に対して変換テーブルに表示される文字の合計数は最高128文字です。



## こんなこともできます

### ■ 文字の読み仮名

単漢字変換は、漢字の音読みを入力して漢字に変換しますが、一部の文字については訓読みにも対応しています。

(例：「川」は、「かわ」「せん」どちらでも変換できます。「木」「髪」も同じ。)

### ■ 単漢字変換モードの開始

単漢字変換モードは **CTRL** + **スペース** のほかに、 **GRAPH** + **SELECT** でも開始・終了できます。

### ■ 読み仮名を入れ直すとき

**BS** キーで1文字分、読み仮名の入力を削除できます。**ESC** キーを押すと、読み仮名を含めて変換テーブルがクリアされます。読み仮名を入力してください。

### ■ 漢字プリンタへの出力

漢字モード(CALL KANJI)でList命令などを実行すると、漢字で出力できます。

### ■ 入力できる文字の種類

文字の種類	大きさ	キー・ランプの状態
ひらがな	全角(かな)	<b>かな</b> ランプのみが点灯。
カタカナ	全角(カナ)	<b>かな</b> 、 <b>CAPS</b> のランプが点灯。単漢字変換モード。
	半角(カタカナ)	<b>かな</b> 、 <b>CAPS</b> のランプが点灯。
記号	全角(●☆)	単漢字変換モードで、 <b>CTRL</b> + 漢字コードを入力。
英数字	全角(A5)	<b>かな</b> ランプが消灯。単漢字変換モード。
	半角(A5B4)	<b>かな</b> ランプが消灯。
漢字	全角(二式)	6ページからの方法で入力します。第二水準の漢字については、漢字ROMを備えていない場合、市販の漢字ROMカートリッジが必要です。

ひらがな、カナ、ローマ字カナ変換については、取扱説明書をご覧ください。

## プログラム中の漢字モード

CALL KANJI命令は、プログラムの中でも実行できます。また、漢字の入力(INPUT命令)も、テキストモードで行うことができます。

### サンプルプログラム

```
10 CALL KANJI
20 INPUT "名前は";A$
30 PRINT A$
40 FOR T=0 TO 200:NEXT T
50 SCREEN 5
60 LOCATE 10,10
70 PRINT A$
80 GOTO 80
```

止めるときは **CTRL** + **STOP** キー

上のプログラムを実行すると、「名前は？」と入力待ちの状態となります。漢字モードとなっていますので、**CTRL** キーを押しながらスペースキーを押して、単漢字変換モードにします。次に、漢字で名前を入力してください。テキスト画面に表示した後、グラフィック画面に名前を表示します。



## プログラムを作るときの注意

漢字を使ってBASICプログラムを作るときの注意点を説明します。

### ■ プログラムの始めにCALL KANJI

CALL KANJI命令は、プログラム中に何回でも実行できます(文字の種類を変えたり、漢字モードを終わることもできます)。最初に使うCALL KANJI命令は、できるだけプログラムの初めの行に入れてください。これは、CALL KANJI命令の実行により、漢字変換用のワークエリア(漢字変換をするために必要なメモリ)を確保するためです。このため、プログラムや変数に使用できるメモリが少し減ります(約550バイト)。また、最初のCALL KANJI命令の実行で、変数、FOR-NEXT、GOSUB命令用のメモリもクリアします。つまり、最初のCALL KANJI命令を実行する前に、変数、FOR-NEXT文、GOSUB命令を使うプログラムを作ってははいけません。

一度ワークエリアを確保すると、2回目以降のCALL KANJI命令の実行では再び確保する必要がないのでメモリのクリアはされません。

### ■ 漢字モードの終了

CALL ANK  $\leftarrow$  (リターンキー)とキー入力すると、通常の英数カナ表示になります。ただし、ワークエリアのメモリは確保した状態です。

プログラム中でも実行できます。

(ANK: 英数カナの意味。Alphabet(英字)、Numeric(数)、Kanaの頭文字をとった用語です)

## ■ MSX, MSX2 のBASIC命令との違い

一部のBASIC命令が漢字モードで使えない、または働きが異なります。また、追加された命令もあります。

命令について詳しくは、20ページからの説明をご覧ください。

BASIC命令	働 き
CALL AKCNV	文字列の中の文字を全角文字列にします。
CALL ANK	漢字モードを終了します。
CALL CLS	画面をクリアします。
CALL JIS	文字列のJISコードを文字列として求めます。
CALL KACNV	文字列の中の文字を半角文字列にします。
CALL KANJI	漢字モードを起動する。
CALL KEXT	文字列の中から半角(全角)文字だけを抜き出します。
CALL KINSTR	文字列の中から指定した文字列を探します。
CALL KLEN	文字列の中の全角(半角)の文字数を得ます。
CALL KMID	文字列の中から指定された位置の文字を抜き出します。
CALL KNJ	文字列で指定された漢字コードの漢字を得ます。
CALL KTYPE	文字列の中の指定された文字のタイプ(全角/半角)を得ます。
CALL PALETTE	カラーパレットの初期化または設定を行ないます。
CALL SJIS	文字のシフトJISコードを得ます。
CLS	画面をクリアします。
COLOR	カラーパレットの初期化または設定を行ないます。
PUT KANJI	SCREEN 5 ~ 12で漢字を表示します。
WIDTH	1行に表示する文字数を設定します。



# 多彩なカラー表示

MSX2+では、MSX2に次の画面表示の機能が追加されています。

- 最大19,268色のカラー表示できる画面モード
- 画面の水平・垂直スクロール

画面の水平・垂直スクロールについては、36ページからのコマンド説明を参照してください。この項では、カラー表示について説明します。

人物や風景などの自然画の再現には、厳密には無限の色数が必要です。このためパソコンで自然画の再現をするとき、表示できる色の数が多いほどリアルな画像となります。

MSX2では最大256色の表示でしたが、MSX2+では最大19,268色の再現が可能です。

## スクリーンの種類

多彩なカラー表現ができる画面モードは、MSX2+で追加された次の3つのモードです(SCREEN 9はありません)。

SCREEN 10: YJK方式による12,499色の表示と、RGB方式の16色(512色中)を同時に表示できます。カラーコードに0~15が指定できます。

SCREEN 11: YJK方式による12,499色の表示と、RGB方式の16色(512色中)を同時に表示できます。SCREEN 10との違いは、カラーコードに0~255が指定でき、RGB方式とYJK方式のデータを扱うことができます。

SCREEN 12: YJK方式による19,268色を同時に表示できます。

YJK方式については、次ページからの説明をご覧ください。

SCREEN 10~SCREEN 12には、SCREEN 8と同じ256×212ドットで画面が構成されます。

## 色の再現方法 - YJK方式とは -


色を再現する方法には、赤・緑・青の色を混合して再現するRGB方式と、明るさ・色あい・色の濃さによって再現する色差方式があります。色差方式は、一般のテレビやビデオで使われている信号の方法です。MSX2+では、パソコン用の色差方式としてYJK方式で色を表現することができます。

### RGB方式について

RGB方式を、SCREEN 8を例に説明します。

SCREEN 8の画面では、画面上の1点(1ドット)がパソコンの1バイトのメモリ(VRAM)に対応します。1バイトは、赤信号3ビット・緑信号3ビット・青信号2ビットの8ビットデータで構成されています。

このため再現できる色の種類は、 $2^3 \times 2^3 \times 2^2 = 256$ 色となります。



RGB方式は、赤・緑・青の明るさを変えることで色を再現します。画面上の1点ごとに直接色をコントロールできるので、プログラムを作るのには便利ですが、多くの色を再現しようとする、パソコンのメモリ(VRAM)が多く必要となります。

### YJK方式について

YJK方式は、画面上の1点ごとに明るさのデータを持ちながら横4ドットの色を1つのデータとして扱います。これは、人間の目の性質として、明るさの違いについては敏感ですが、色の違いについては感度が低いことを利用しています。つまり、色の付け方はおおまかでも、明るさが細かく表現できればきれいに見えることとなります。

これにより、画面の解像度はSCREEN 8と同じ256×212ドットでありながら、原理的に約13万色( $2^5 \times 2^6 \times 2^6 \approx 13$ 万)を表示することができます(実際には、目に見える色の違いとして19,268色)。




## YJK方式のデータ(SCREEN 12の場合)

YJK方式は横256ドット×縦212ドットの画面モードで、横方向に連続した4ドットを一組として色を再現します。

画面上の連続した4ドットのデータの内容は、次のように構成されています。

D<sub>7</sub> D<sub>6</sub> D<sub>5</sub> D<sub>4</sub> D<sub>3</sub> D<sub>2</sub> D<sub>1</sub> D<sub>0</sub>



(SCREEN 12のとき)

Yは明るさ(輝度)のデータです。K、Jは色あいのデータです。K<sub>L</sub>はKの下位3ビット、K<sub>H</sub>はKの上位3ビットです。Jも同じです。

各ドットごとに独立してYのデータを5ビット持ち、横4ドット共通としてKのデータを6ビット、Jのデータを6ビット持っています。これにより、一つの点の色はY(5ビット)×K(6ビット)×J(6ビット)の組合わせで再現されます。

各点のYJKのデータから画面に再現されるRGBの色への変換は、パソコンのVDPの内部で次の処理を行います。

$$R = Y + J$$

$$G = Y + K$$


$$B = \frac{5}{4} Y - \frac{1}{2} J - \frac{1}{4} K$$

Yの値は0～31(実際はD<sub>3</sub>からのデータだから3ビットシフトして、8の倍数である0, 8, 16, ..., 248の値)です。

JおよびKは符号付きで処理され、値は-32～+31です。(実際にLINE命令などで指定するときは0～63です。)

### <YJKの色の概念図>

注意:この図はイメージをつかむための図です。実際の色とは異なります。



## YJK方式のしくみ

次のプログラムを入力してください。

SCREEN 12の19,268色を全て表示させることは簡単にはできませんが、YJK方式によるカラーパターンを再現しています。

```


100 COLOR ,1,15:SCREEN 12
110 '***** Y Set *****
120 FOR Y=0 TO 106 STEP 3
130   CC=VAL("&B"+BIN$(Y/3.3)+"000")
140   LINE(0,Y)-(255,Y+2),CC,BF
150   LINE(0,211-Y)-(255,209-Y),CC,BF
160 NEXT Y
170 '***** K Set *****
180 FOR Z=0 TO 128 STEP 64
190   FOR X=0 TO 63 STEP 4
200     K=X:K1=K AND 7:K2=K&8
210     LINE(X+Z,0)-(X+Z,211),K1,BF,OR
220     LINE(X+1+Z,0)-(X+1+Z,211),K2,BF,OR
230     IF INKEY$<>"" THEN STOP
240   NEXT X,Z
250 '***** J Set *****
260   Z=61: FOR X=0 TO 63 STEP 4
270     J=X:J1=J AND 7:J2=J&8
280     LINE(63-X+Z+2,0)-(63-X+Z+2,211),J
290     LINE(63-X+3+Z,0)-(63-X+3+Z,211),J
300     IF INKEY$<>"" THEN STOP
310   NEXT X
320 FOR Z=128 TO 255 STEP 64
330   FOR X=0 TO 63 STEP 4
340     J=X:J1=J AND 7:J2=J&8
350     LINE(X+Z+2,0)-(X+Z+2,211),J1,BF,OR
360     LINE(X+3+Z,0)-(X+3+Z,211),J2,BF,OR
370     IF INKEY$<>"" THEN STOP
380   NEXT X,Z
390 GOTO 390

```


Yデータを  
上下に変化させま  
す。

Kデータを左から  
変化させます。

Jデータを左右に  
変化させます。



<実行画面>





次のプログラムは、前ページのプログラムで描いたパターンの指定した位置の Y J K データを表示させます。2つのプログラムを組み合わせると Y J K の値と表示される色の関係がわかります。


17ページのプログラムの390行からを入力し直してください(このプログラムの380行は単独でも動作するためです。組み合わせるときは必要ありません)。

```
380 SCREEN 12
390 X=125:Y=106
400 OPEN "GRP:" AS #1
410 GOSUB 620:GOSUB 480
420 '** YJK ケイザン **
430 C1=POINT(X,Y):YD=C1/8:XX=(X/4)*4
440 K1=POINT(XX,Y) MOD 8:K2=POINT(XX+1,Y
) MOD 8
450 J1=POINT(XX+2,Y) MOD 8:J2=POINT(XX+3
,Y) MOD 8
460 K=K1+K2*8:J=J1+J2*8
470 GOTO 410
480 '** INPUT check & Sprite **
490 SPRITE$(0)=STRING$(2,16)+CHR$(0)+CHR
$(198)+CHR$(0)+STRING$(2,16)
500 A=STICK(0)
510 IF A=1 THEN Y=Y-1
520 IF A=2 THEN Y=Y-1:X=X+1
530 IF A=3 THEN X=X+1
540 IF A=4 THEN Y=Y+1:X=X+1
550 IF A=5 THEN Y=Y+1
560 IF A=6 THEN Y=Y+1:X=X-1
570 IF A=7 THEN X=X-1
580 IF A=8 THEN Y=Y-1:X=X-1
590 PUT SPRITE 0,(X-3,Y-3),8
600 IF STRIG(0)=0 THEN 500
610 RETURN
620 '** TABLE ヒョウジ **
630 LINE(12,168)-(79,200),0,BF,PSET:COLO
R 248,0
640 PSET (20,169),0:PRINT #1,USING "Y=##
";YD
650 PSET (20,179),0:PRINT #1,USING "K=##
";K
660 PSET (20,189),0:PRINT #1,USING "J=##
";J
670 RETURN
680 '** END ***
690 SCREEN 0:END
```

16、17ページのプログラムを組合わせて実行すると、右の図のような画面となります。

カーソルキーで移動させます

カーソルキーで“+”のスプライトを表示させたい色の位置に移動させ、スペースキーを押してください。画面の左下に、YJKの値が表示されます



**ちょっと一言**

**■ Y成分による色の変化**

16ページのプログラムを実行して、写真のように画面が表示されましたか？

画面を見て、縦の方向で色が違って見えますね？

画面の縦方向では、Y成分つまり明るさしか変わらないはずなのに、色が違って見えます。例えば右側の赤の色でも、明るければ白っぽく(黄色に)見え、暗ければ黒っぽく見えます。

これは14ページにあるように、人の目が明るさに対して、色合いについては感度が低いことを意味します。



## YJK方式のデータ(SCREEN 10、11の場合)

SCREEN 10と11ではYのデータを4ビットにし、8バイト中の1ビット(D<sub>3</sub>)を属性(アトリビュート)ビットとしています。

この方式では、属性ビットが0の場合、その点はYJK方式で表示されます。属性ビットが1の場合、その点はYの4ビットのデータがカラーコードとして扱われます。このカラーコードはカラーパレット機能を使って512色中16色をRGBのデータとして表示することができます。

D<sub>7</sub> D<sub>6</sub> D<sub>5</sub> D<sub>4</sub> D<sub>3</sub> D<sub>2</sub> D<sub>1</sub> D<sub>0</sub>

Y <sub>1</sub>	A <sub>1</sub>	K <sub>L</sub>
Y <sub>2</sub>	A <sub>2</sub>	K <sub>H</sub>
Y <sub>3</sub>	A <sub>3</sub>	J <sub>L</sub>
Y <sub>4</sub>	A <sub>4</sub>	J <sub>H</sub>

(SCREEN 10, 11のとき)

Yは明るさ(輝度)のデータです。K、Jは色あいのデータです。K<sub>L</sub>はKの下位3ビット、K<sub>H</sub>はKの上位3ビットです。Jも同じです。

Yの値は0～15(実際は4ビットシフトして、0, 16, 32, …, 240の値)です。JおよびKは符号付きで処理され、値は-32～+31です。(実際にLINE命令などで指定するときは0-63です。)

SCREEN 10の場合、YJK方式の画面にRGB方式で図を描くとき、LINE命令などの論理演算子の指定により、Yの値を意識しないで0～15のカラーコードで描画色を指定できます。カラーコードの値はY成分にしか影響しません。また、描画した場合、属性ビットが1にセットされます。

SCREEN 11の場合、YJK方式の画面にRGB方式で図を描くとき、カラーコードとして0～255が指定できます。Y成分に加えてJK成分の値も変えることができます。YJK方式の画面の色を変えたくないときは、下位3ビットを考慮してカラーコードを指定してください。

SCREEN 10、11は、RGB方式で描画した場合、描画したカラーコードに対してカラーパレット機能が使用できます。

# MSX<sub>2+</sub>のコマンド -MSX,MSX<sub>2</sub>との違い-

MSX<sub>2+</sub>で使う命令や関数などを、アルファベット順に説明しています。またMSX、MSX<sub>2</sub>で使っている命令でも、MSX<sub>2+</sub>の機能を使ったときに使い方が変わる命令については、相違点のみ説明しています。

MSX、MSX<sub>2</sub>の命令については、別冊の「MSXプログラミング入門」をご覧ください。

## 機能別索引

\*印のコマンドは、MSX、MSX<sub>2</sub>から機能が変わります。

### 画面の制御・関数

#### 関数・アドレス

テーブルのアドレス を得る	BASE*.....22	VDPレジスタの内 容を読み出す/代入 する	VDP*.....37
------------------	--------------	------------------------------	-------------

#### 画面の制御

漢字画面をクリアす る	CALL CLS ...23	画面をスクロールす る	SET SCROLL ....36
画面をクリアする	CLS*.....32	VDPレジスタの内 容を読み出す/代入 する	VDP*.....37
画面モードを指定す る	SCREEN*.....34	表示する文字数の設 定	WIDTH*.....38

#### 色の制御

カラーパレットを変 える	CALL PALETTE ...31	画面の色を指定する	COLOR*.....32
-----------------	-----------------------	-----------	---------------



## 漢字の出力・制御

### 漢字表示の制御





漢字モードを終了します	CALL ANK ……23	漢字の表示位置を指定する	LOCATE* ……33
漢字モードを起動します	CALL KANJI ……26	漢字を表示します	PUT KANJI* ……33

### 文字列の制御

文字列を全角に変える	CALL AKCNV ……22	文字列の指定した位置から文字列を取り出す	CALL KMID ……29
文字の J I S コードを求める	CALL JIS ……24	文字列で指定された漢字コードの漢字を得る	CALL KNJ ……30
文字列を半角に変える	CALL KACNV ……25	文字列の指定した位置の文字の半角／全角を知る	CALL KTYPE ……30
文字列の半角／全角文字だけを抜き出す	CALL KEXT ……28	文字列のシフト J I S コードを得る	CALL SJIS ……31
文字列の中の指定した文字の位置を知る	CALL KINSTR ……28		
文字列の長さを知る	CALL KLEN ……29		





# BASE\*

Base(ベース:基本)

- 
- 働き**  画面出力に関連するテーブル(VDPレジスタの各テーブル)の先頭アドレスを得ます。
- 書き方**  BASE(<数式>)
- 例** PRINT BASE(10)  
6144
- 説明**  画面表示を行うVDPレジスタ内の各テーブルの先頭アドレスを得ます。指定できる数値は0~44および50~64です。  
このシステム変数に値を代入することはできません。
- 参照**  VDP (37ページ)

# CALL AKCNV





ANK Kanji Conversion(アंक・カンジ・コンバージョン:ANK文字を漢字に変換する)

- 
- 働き**  文字列の中の文字を全角文字列にします。
- 書き方**  CALL AKCNV(<文字変数>,<文字列>)
- 例** CALL AKCNV(A\$,"ABC漢字イロハ"):PRINT A\$  
ABC漢字イロハ
- 説明**  <文字列>中の全ての文字を全角文字に変換して<文字変数>に代入します。  
ANK文字(Alphabet-Numeric-Kana文字)を半角文字と言います。  
半角文字と漢字(全角文字)の混じった文字列を全て全角文字に置き換えることができます。
- 参照**  CALL KACNV (25ページ)







# CALL ANK

Alphabet-Numeric-Kana(アंक:英数カナ文字。半角で表すことのできる文字)

- 
- 働き  漢字モードを終了します。
- 書き方  CALL ANK
- 説明  漢字表示を終了します。ただし漢字処理用に確保したシステムのメモリは開放されません。  
漢字表示を終了すると、漢字で作成されたプログラムなどは漢字表示されなくなります。また、プリンタへのリスト出力なども漢字対応でなくなります。
- 参照  CALL KANJI (26ページ)





# CALL CLS

Clean Screen(クリーン・スクリーン:画面のクリア)

- 
- 働き  画面をクリアします。
- 書き方  CALL CLS
- 説明  CALL KANJI 命令により画面を漢字モードにしたとき、画面をクリアするために使用します。漢字モードでないときでも使用できます。  
漢字モードでCLS命令を実行すると、“Illegal function call”エラーとなります。ただし、グラフィック画面(SCREEN 2 ~ 12)では、CLS命令も使用できます。
- 参照  CLS (32ページ)

# CALL JIS





JIS(ジス:JIS規格の文字コード)

- 
- 働き**  文字列のJISコードを文字列として求めます。
- 書き方**  CALL JIS(<文字変数>,<文字列>)
- 例** CALL JIS(A\$,"漢字"):PRINT A\$  
3 4 4 1
- 説明**  <文字列>の最初の2バイトを16進4桁のJISコードに変換して<文字変数>に代入します。
- 参考** 文字の表し方には、JISコードとシフトJISコードがあります。  
JISコードは日本工業規格(Japanese-Industrial-Standards)のルールです。  
JISコードは、漢字のデータの始まりに制御命令(SIコード)を入れ、データの終りにも制御命令(SOコード)を入れます。このため、漢字とANK文字が混じると、メモリが多く必要となります。  
これに対して、シフトJISコードは、JISコードで決められていないコードに漢字制御と漢字の意味を持たせ、2バイトで漢字の表示を行います。メモリは少なくてすみませんが、パソコン独自のグラフィック記号(●や♣など)は使えなくなります。  
16ビットのパソコンなどには、シフトJISコードで動作するソフトウェアもあり、データを交換するときに変換が必要となります。
- 参照**  CALL SJIS (31ページ)



# CALL KACNV

Kanji ANK Conversion(カンジ・アंक・コンバージョン:漢字をANK文字に変換する)


- 
- 働き  文字列の中の文字を半角文字列にします。
- 書き方  CALL KACNV (<文字変数>, <文字列>)
- 例 CALL KACNV(A\$, "私はプログラマーです"):PRINT A\$  
私ハプログラマーデス
- 説明  <文字列>中の半角文字に変換できる文字(英数字、カタカナ、ひらがな)を全て半角文字に変換して<文字変数>に代入します。
- 参照  CALL AKCNV (22ページ)

# CALLKANJI

Kanji(カンジ:漢字モードの開始)

働き 

漢字モードを起動する。


書き方 

CALLKANJI (<漢字モード>)

例

CALLKANJI

CALLKANJI1

説明 

漢字表示モードは、0, 1, 2, 3のどれか1文字で、表示される漢字の大きさ(表示方法)を指定します。省略すると0と見なされます。6ページからの説明もご覧ください。

- CALLKANJI (またはCALLKANJI0)

漢字の文字パターン(フォントデータ)を漢字ROMから取り出して表示します。文字サイズは全角で横16ドット×縦16ドットです(カナ、英数字は半角で8×16ドット表示)。

- CALLKANJI1

文字パターン(フォントデータ)を漢字ROMから取り出し、横16ドットを12ドットに圧縮して表示します(12ドットの漢字ROMが差し込まれている場合は、このROMから表示します)。文字サイズは横12ドット×縦16ドットです。

- CALLKANJI2

CALLKANJI0と同じですが、画面表示がインターレースモードとなり、縦方向の表示文字数(行数)が2倍となります。文字パターンは16×16ドットですが、見た目には16×8ドットとなります。

- CALLKANJI3

CALLKANJI1と同じですが、画面表示がインターレースモードとなり、縦方向の表示文字数(行数)が増します。BASICのコマンド待ちの状態(テキストモード)では、漢字の入力・表示が可能です。SCREEN2~12のグラフィックモードではプログラムによる漢字の出力のみが可能です。

日本語処理機能(MSX-JE)を持つカートリッジが差し込まれているときは、CALLKANJI命令の実行時にシステムに組み込まれます。このときは、連文節変換が可能です。



プログラム中で最初のCALL KANJI命令を実行するときに、漢字変換用のワークエリア(漢字変換をするために必要なメモリ)を確保します。このため、プログラムや変数に使用できるメモリが少し減ります(約550バイト)。また変数、FOR-NEXT、GOSUB命令用のメモリもクリアされます。一度ワークエリアを確保すると、2回目以降のCALL KANJI命令の実行ではメモリのクリアはされません。

なお、漢字モードを終了(CALL ANK命令を実行)してもメモリは確保されたままの状態です(リセットされるまで解放しません)。


## 参 考

### 漢字モードのときの表示幅

CALL KANJI実行後の表示幅は、命令実行前の表示幅(ANKモードの表示幅)から次のように決定されます。




- SCREEN 0 を使用していたとき
- KANJI 0、KANJI 2 命令のとき  
ANKモードの表示幅の $\frac{4}{5}$ が漢字モードの表示幅になります。
- KANJI 1、KANJI 3 命令のとき  
ANKモードの表示幅がそのまま漢字モードでの表示幅になります。
- SCREEN 1 を使用していたとき
- KANJI 0、KANJI 2 命令のとき  
ANKモードの表示幅がそのまま漢字モードでの表示幅になります。
- KANJI 1、KANJI 3 命令のとき  
ANKモードの表示幅の $\frac{5}{4}$ が漢字モードの表示幅になります。

ただし、計算によって決まった表示幅が20未満の時は20に設定されます。また、グラフィックモードにしたときは、テキストモードの表示幅に関係なく、常に表示できる最大幅に設定されます。

参 照  WIDTH (38ページ)




# CALL KEXT

Kanji Extract(カンジ・イクストラクト:漢字を抜き取る)

- 
- 働き  文字列の中から半角(全角)文字だけを抜き出します。
- 書き方  CALL KEXT(<文字変数>, <文字列>, <機能>)
- 例 CALL KEXT(A\$, "今日は良天気.", 0):PRINT A\$  
1.  
CALL KEXT(A\$, "今日は良天気.", 1):PRINT A\$  
今日は良天気
- 説明  <機能>が0なら<文字列>中の半角文字(ANK文字)だけを、1なら全角文字だけを抜き出し<文字変数>に代入します。

# CALL KINSTR




Kanji in String(カンジ・イン・ストリング:文字列の中の漢字)

- 
- 働き  文字列の中から指定した文字列を探します。
- 書き方  CALL KINSTR(<数値変数>, [<数式>], <文字列1>, <文字列2>)
- 例 CALL KINSTR(A, "A亜B", "B"):PRINT A  
3
- 説明  <文字列1>の中から<文字列2>を探しだし、見つければ発見した位置を、見つからなければ0を<数値変数>に代入する。  
<数式>は探し始める位置を文字数で指定します。省略されると1とみなされる。  
探し出す文字列は、全角/半角を区別して探します。  
また全角文字の1文字も1文字として扱われます。






# CALL KLEN

Kanji Length(カンジ・レングス:漢字の長さ)

- 
- 働き**  文字列の中の全角(半角)の文字数を得ます。
- 書き方**  CALL KLEN(<数値変数>, <文字列>, [<機能>])
- 例**
- ```
CALL KLEN(A, "今日ハ良イ天気です"):PRINT A
9
CALL KLEN(A, "今日ハ良イ天気です", 1):PRINT A
2
CALL KLEN(A, "今日ハ良イ天気です", 2):PRINT A
7
```
- 説明**  機能が0(もしくは省略)の時は<文字列>の全体の文字数を、1の時は<文字列>中の半角文字の文字数を、2の時は<文字列>中の全角文字の文字数を<数値変数>に代入します。


# CALL KMID


Kanji Middle(カンジ・ミドル:漢字文字列の中)

- 
- 働き**  文字列の中から指定された位置の文字を抜き出します。
- 書き方**  CALL KMID(<文字変数>, <文字列>, <式1>, [<式2>])
- 例**
- ```
CALL KMID(A$, "今、何時?", 3, 2):PRINT A$
何時
```
- 説明**  <文字列>中の<式1>番目の文字から<式2>文字分だけ抜き出して<文字変数>に代入します。<式2>が省略された場合は<式1>番目の文字から終わりまで全ての文字が代入されます。

# CALL KNJ


Kanji(カンジ:漢字)

働き  文字列で指定された漢字コードの漢字を得ます。

書き方  CALL KNJ (<文字変数>, <文字列>)

例 CALL KNJ (A\$, "3441"):PRINT A\$


漢

説明  <文字列>で指定される4桁の漢字コードに相当する漢字1文字を<文字変数>に代入します。漢字コードが&H8000未満の時はJISコード、以上の時はシフトJISコードとみなされます。

参照  CALL JIS (24ページ)

# CALL KTYPE


Kanji Type(カンジ・タイプ:漢字の形)

働き  文字列の中の指定された文字のタイプ(全角/半角)を得ます。

書き方  CALL KTYPE (<数値変数>, <文字列>, <数式>)

例 CALL KTYPE (A, "今日ハ", 2):PRINT A

1

説明  <文字列>中の<数式>番目の文字のタイプ(半角なら0、全角なら1)を<数値変数>に代入します。

<数式>は省略できません。<数式>を省略したり、文字列の文字数より大きい数を指定するとエラーとなります。

例).

```




1 0 A$ = "今日ハ良い天気です"
2 0 CALL KLEN(L, A$)
3 0 FOR I = 1 TO L
4 0 CALL KTYPE(T, A$, I):PRINT T;
5 0 NEXT I
RUN
1 1 0 1 0 1 1 1 1

```







# CALL PALETTE

Palette(パレット:調色板、色影)

- 
- 働き**  カラーパレットの初期化または設定を行ないます。
- 書き方**  CALL PALETTE(〈カラーコード〉,〈赤レベル〉,  
〈緑レベル〉,〈青レベル〉)
- 例** CALL PALETTE(15,4,4,4)
- 説明**  CALL KANJI 命令により画面を漢字モードにしたときにパレットを変更するために使用します。漢字モードでないときでも使用できます。
- 漢字モードでは、COLOR 命令で色レベルを変更しようとする  
と、"Illegal function call"エラーとなります。
- なお、グラフィック画面(SCREEN 2~12)では、COLOR  
R 命令で色レベルを変更することもできます。
- パラメータをすべて省略した場合は、パレットを初期状態にしま  
す。
- パラメータが指定された場合、〈カラーコード〉で指定された色を  
〈赤レベル〉,〈緑レベル〉,〈青レベル〉の色に設定します。なお、カ  
ラーコードや各レベルを省略することはできません。





# CALL SJIS

Shift JIS(シフト・ジス:移動JISコード)

- 
- 働き**  文字のシフトJISコードを得ます。
- 書き方**  CALL SJIS(〈文字変数〉,〈文字列〉)
- 例** CALL SJIS(A\$, "漢字"):PRINT A\$  
8ABF
- 説明**  〈文字列〉の最初の2バイトを16進4桁のシフトJISコードに  
変換して〈文字変数〉に代入します。
- 参照**  CALL JIS (24ページ)





# CLS\*

Clean Screen(クリーン・スクリーン:スクリーンのクリア)

- 
- 働き  画面をクリアします。
- 書き方  CLS
- 説明  CLS命令は、CALL KANJI命令により画面を漢字モードにすると、“Illegal function call”となり使用できません。漢字モードの時はCALL CLS命令を使用してください。ただし、グラフィックモード(SCREEN 2~12)では使用できません。
- 参照  CALL CLS (23ページ)

# COLOR\*




Color(カラー:色)


- 
- 働き  カラーパレットの初期化または設定を行ないます。
- 書き方  COLOR  
COLOR=NEW  
COLOR=RESTORE  
COLOR=(〈カラーコード〉,〈赤レベル〉,〈緑レベル〉,  
〈青レベル〉)
- 説明  CALL KANJI命令により漢字モードにすると、COLOR命令のカラーパレット機能は、“Illegal function call”となり使用できません。漢字モードの時はCALL PALETTE命令を使用してください。ただし、グラフィックモード(SCREEN 2~12)では使用できません。
- 参照  CALL PALETTE (31ページ)



# LOCATE\*




Locate(ロケイト:置く、設定する)

- 働き  テキスト画面のカーソルの位置および動きを指定します。
- 書き方  LOCATE(<X座標>),(<Y座標>),(<カーソルスイッチ>)
- 例 LOCATE 15, 10
- 説明  カーソルをテキスト画面の指定した位置(X,Y)へ移動します。この命令は、PRINT文で表示する文字やINPUT文で表示するプロンプト文の開始位置を指定するときなどに用います。(入力待ちのときの?の位置を指定できます。)
- <X座標>は、画面の右端を0とする水平座標を表す数で、32×24テキストモードのとき0~31、40×24テキストモードのとき0~39です。省略した場合は0とみなされます。
- CALL KANJI命令で画面を漢字モードにした後、グラフィックモード(SCREEN 2~12)の画面に、PRINT命令で文字を表示させることができます。このとき、表示位置をLOCATE命令で指定することができます。ただし、カーソルスイッチは指定できません。
- <X座標>の値はテキストモードの表示幅に関係なく、グラフィック画面で表示できる最大文字数までの値となります。
- 例. SCREEN 5、CALL KANJIのとき、  
<X座標>は0~32です。

参照  CALL KANJI (26ページ)



# PUT KANJI\*

Put Kanji(フット カンジ:漢字を出力する)

- 働き  SCREEN 5から12で漢字を表示します。
- 書き方  PUT KANJI ((<X>,<Y>)),<漢字コード>[,<カラーコード>][,<ロジカルオペレーション>][,<モード>]
- 例 PUT KANJI (100,100),&H3021
- 説明  MSX2+では、漢字コードがJIS第二水準まで指定できるようになりました。ただし、表示するためにはJIS第二水準漢字ROMが必要となります(MSX2+では、第二水準の漢字ROMはオプション装備です)。


# SCREEN\*

Screen(スクリーン:画面)

働き  画面やスプライトの大きさなどについての設定を行ないます。  
書き方  SCREEN(<画面モード>)[,<スプライトサイズ>][,<キークリックスイッチ>][,<カセットボーレート>][,<プリンタオプション>][,<インターレースモード>]

例

SCREEN 12

説明 

MSX2+にはMSX2に、SCREEN10~12の画面モードが追加されています。なお、SCREEN9はありません。<画面モード>以外の使用方法は、MSX2と同じです。<画面モード>以外については、別冊の「MSXプログラミング入門」をご覧ください。また、YJK方式については、13ページの説明もご覧ください。

## • SCREEN 10

属性(アトリビュート)混在のYJKモードを選択します。このモードで指定できるカラーコードの範囲は0~15です。描画のときの処理は、どんな論理演算子を指定したかにより変わります。

### • 論理演算子がPSETまたは省略されたとき

PSETやLINE命令で描画するとき、描いた点の属性に1がセットされ、上位の4ビット(Y成分のデータ)が、指定されたカラーコードとなります。カラーコードはCOLOR命令などのパレット機能により表示する色を512色中16色で指定できます。

### • 論理演算子にPSET以外が指定されたとき

指定されたカラーコードと、描画する場所のY成分とのあいだで論理演算が実行されます。カラーコードは、属性が1にセットされている点に対してはパレット機能を持つカラーコードとして指定していることになり、属性がセットされていない点に対してはY成分として指定することになります。

## • SCREEN 11

属性混在のYJKモードを選択します。このモードで指定できるカラーコードの範囲は0~255です。

SCREEN10と違い、描画のときに指定されたカラーコードは、そのまま描画する点のYJK成分と論理演算されます。このため、Y成分のほかにJK成分も変わり、描画した点の色が変わることもあります。描画した結果、属性が1にセットされた点のカラーコードに対して、パレット機能で表示する色を変えることもできます。



---

- SCREEN 12

属性なしのYJKモードを選択します。このモードで指定できるカラーコードの範囲は0～255です。

描画のときに指定されたカラーコードは、そのまま描画する点のYJK成分と論理演算されます。

MSX、MSX2のSCREEN文では、〈画面モード〉が変わると表示を初期化(画面のクリア)しましたが、SCREEN10,11,12の間での切換えでは次のようになります。

- 画面表示がクリアされない画面モード切換え

- ・ SCREEN 10 ⇒ SCREEN 11
- ・ SCREEN 10 ⇒ SCREEN 12
- ・ SCREEN 11 ⇒ SCREEN 10
- ・ SCREEN 11 ⇒ SCREEN 12

- 画面表示はクリアされませんが、属性が0にされる画面モード切換え (Y成分が5ビットから4ビットになるため、階調が下がります)

- ・ SCREEN 12 ⇒ SCREEN 10
- ・ SCREEN 12 ⇒ SCREEN 11

### サンプルプログラム

SCREEN 10～12の画面に、LINE命令で線を描きます。150行の画面モードと、160行の論理演算子(OR,AND,PSET)を変えてみてください。


```

100 COLOR ,1,15:SCREEN 12
110 FOR Y=0 TO 211 STEP 3
120   CC=VAL("&B"+BIN$(Y/7)+"000")
130   LINE(0,Y)-(255,Y+2),CC,BF
140 NEXT Y
150 SCREEN 11
160 LINE(0,0)-(255,211),8,,OR
170 GOTO 170


```

# SET SCROLL

Set Scroll(セット スクロール:巻き物、巻く)

働き 


表示画面を水平及び垂直方向にスクロールします。

書き方 

SET SCROLL(<X>)[,<Y>][,<マスク>][,<2 ページ>]

例

SET SCROLL 128,100,0,0

説明 

表示画面を水平および垂直方向にスクロールします。<X>で水平方向(左へ)の、<Y>で垂直方向(上へ)のスクロール量を数式で指定します。<X>の範囲は0~511、<Y>の範囲は0~255です。

<X>、<Y>の値は、始めの位置からいくらスクロールするかの値です。現在の位置からのスクロール値ではありません。

<マスク>は表示画面の左端8ドットを周辺色で隠すかどうかを指定します。これは、水平スクロールのとき、画面の左端に画面の切れが見えるためです。<マスク>が0の時は隠しません。<マスク>が1の時は隠します。

<2 ページ>はスクロールをしたときに、表示されている画面(ページ)の右端または左端に次のページを表示するかを数で指定します。<2 ページ>が0ならばスクロールをしたときに同じページ内で表示が繰り返されます。<2 ページ>が1ならば2ページ連続のスクロールになります。

例えば、表示しているページが左へ移動すると、移動した量だけ次ページの画面の内容が、右端に表示されます。また、この時はあらかじめSET PAGE命令で表示ページを奇数ページにする必要があります。

注意

このSET SCROLL命令はSCREEN 0の時は動作が異なります。

表示されている文字の位置は変更されず、文字のフォントの表示開始位置が変わります(文字の形がくずれます)。

水平スクロール量は<X>の下位3ビットのみが有効となり、1を指定すると7ドット右にスクロールします。2の時は6ドット、7の時は1ドットとなります。7ドット以上はスクロールしません。

垂直スクロール量は<Y>の下位3ビットのみが有効となり、1を指定すると1ドット上にスクロールします。2の時は2ドット、7の時は7ドットとなります。7ドット以上はスクロールしません。



---

 サンプルプログラム

35ページのプログラムを変更して、水平・垂直スクロールを行います。


```

100 COLOR ,1,15:SCREEN 12
110 FOR Y=0 TO 211 STEP 3
120   CC=VAL("&B"+BIN$(Y/7)+"000")
130   LINE(0,Y)-(255,Y+2),CC,BF
140 NEXT Y
150 SCREEN 11
160 LINE(0,0)-(255,211),8,,OR
170 COPY (0,0)-(255,50) TO (0,212)
180 FOR X=0 TO 255
190 SET SCROLL X,X,1
200 NEXT X:GOTO 180
  
```


## VDP\*

VDP(ブイ・デー・ピー)

---

 働き 


VDPのレジスタへデータを書き込んだり、レジスタからデータを読みだしたりします。

書き方 

VDP(<レジスタ番号>)

## 例

VDP(26)=VDP(26)OR&H8

説明 


VDP(ビデオ・ディスプレイ・プロセッサ)のレジスタへデータを書き込んだり、レジスタからデータを読みだしたりします。

指定できる<レジスタ番号>は-9~-1、0~28および33~47です。-9~-1がVDPのステータスレジスタ9~1へ、0~7がレジスタ0~7へ、8がステータスレジスタ0へ、9~28がレジスタ8~27へ、33~47がレジスタ32~46へ対応しています。


この変数は、VDPについての知識を得てから使用してください。

# WIDTH\*

Width(ウイズ:広さ、幅)

働き 


画面に表示する文字数を設定します。

書き方 

WIDTH<桁数>

例

WIDTH 28

説明 

WIDTH命令はCALL KANJI命令を実行したときには次のように動作します。

- テキストモードの場合 (SCREEN 0と1)

- KANJI 0かKANJI 2の場合  
 <桁数>の指定は20から64が有効です。
- KANJI 1かKANJI 3の場合  
 <桁数>の指定は20から80が有効です。

上記以外の値が指定されると "Illegal function call" となります。

- グラフィックモードの場合 (SCREEN 2~12)

常に "Illegal function call" となります。

漢字表示を行なわないときは、MSX2と同じ働きをします。



# MSX-MUSICの使い方



この章では、MSX2+のオプション機能であるMSX-MUSICの機能について説明します。

MSX-MUSICにより、FM音源を使ってピアノ、ギター、オルガンなど、64種類の音色を出すことができます。また、8オクターブ9重和音の演奏もできます。

音楽の演奏は、MSX BASICの機能を拡張したBASIC命令により簡単にできます。あなたも、コンピュータミュージックに挑戦してください。

FM音源とは? ..... 40

• サンプルプログラム ..... 40

音楽命令の使い方 -コマンド一覧- ..... 42

注意：MSX-MUSICが使えるパソコン

パソコン本体に $\square$ マークがついていると、MSX-MUSICの機能が使えます。

# F M音源とは？

別冊の「MSXプログラミング入門」では、PLAY命令とSOUND命令を使った音の出し方が説明してあります。PLAY命令とSOUND命令は、MSX、MSX2標準のPSG(プログラマブル・サウンド・ゼネレータ)を使って音を出します。

この章で説明するMSX-MUSICは、MSX2+のオプション機能として追加されています。MSX-MUSICは、電子楽器(シンセサイザー)などに使われているFM音源と同じ原理で音を出すことができ、PSGを使った音に加えて8オクターブ9重和音(または6重和音+5リズム音)を発生できます。また、☐マークのついたゲームを使うと、ゲームの音が迫力ある音になります。

## サンプルプログラム

☐マークのついたゲームを持っていないとき、また簡単に音の違いを知りたいとき、次のプログラムを実行してみましょう。3重和音のFM音源でオルゴールの音を出します。

120行、150行のPLAY命令の「#2」を「#0」にするとPSGの音(3重和音)となります。

また、120行の「T90」を「T90@n」(nは0~63の数字)に変えると、音の種類を変えることができます。

```
100 CALL MUSIC(0,0,1,1,1)
110 CALL TRANSPOSE(2400)
120 PLAY #2,"T90","T90","T90"
130 READ A$:IF A$="END" THEN RESTORE:GOTO 130
140 READ B$,C$
150 PLAY #2,A$,B$,C$
160 GOTO 130
170 DATA L16,L16,L16
180 DATA 05C8DC04B05CA4F8
190 DATA 04F4F805C404A8
200 DATA 03A8B-AG+A04F4C8
210 DATA 05E4D8L4G.L16
220 DATA A4G8L4B-.L16
230 DATA 03B-4B-804L4D.L16
240 DATA L804B-05C.D16EF.G16
250 DATA L804DF.F16GB-.B-16
260 DATA L803GA.B-1604CD.E16
270 DATA END
```



## ■音楽用LSIについて

MSXパソコンで音を出すには、次の3つの方法があります。

- MSX標準のPSG (プログラマブル・サウンド・ジェネレータ)を使う。
- MSX2+ のMSX-MUSIC (オプション装備)を使う。
- MSX-AUDIO (カートリッジなど)を使う。

PSGでは、ノイズ発生器で音の高さに合わせて一定の周波数の信号を作り、この信号の出力する大きさと、出力する間隔を変えることで、ピアノやオルガンに似た音を作っています。しかし、実際の楽器の音は何種類もの(周波数の)音の合成でできているので、一定の信号ではリアルな音になりません。

MSX2+ とMSX-AUDIOでは、FM音源(FM方式の音出力)を採用しています。PSGと異なり、出力する信号の周波数を変えることができます。


MSX2+ の音楽用LSIは、MSX-AUDIOの演奏方式をパソコンに内蔵できるように小型化し、機能の一部を変更したものです。次の特長があります。

- FM音源を使用しています。
- 演奏のモードを選択することで9重和音またはメロディー6音+リズム5音の演奏ができます(CALL MUSIC命令で選択)。
- 音色データ(音色番号0~63中の15種類)をROMとして持つので、音色の変更が簡単にできます。また、独自の音を作れるように、1音分のデータ領域(音色番号63)があります。

## ■FM方式とは

音の信号を一定の信号(ノイズ)で作る方法をAM方式といい、周波数を変えて作る方法はFM方式(Frequency Modulation:周波数変調)といいます。

FM方式は、信号の周波数を変えることができるので、自然楽器の合成音から電子音までの音を作ることができます。



# 音楽命令の使い方 -コマンド一覧-

## 機能別索引

\*印のコマンドは、MSX、MSX2から機能が変わります。

### 関数・レジスタ制御

レジスタに書き込み します	CALL AUDREG ..43	演奏中かどうかを調 べます	CALL PLAY ..48
------------------	------------------	------------------	----------------

### 演奏の制御

バックグラウンド処理 を設定します	CALL BGM ....44	バックグラウンド処理 を中止します	CALL STOPM..48
システムの初期化	CALL MUSIC ..45	演奏します	PLAY*.....56




### 音の制御

音の高さを変えます	CALL PITCH ..47	音色を設定します	CALL VOICE ..51
音律を設定します	CALL TEMPER ....49	音色をコピーします	CALL VOICE COPY ...54
移調を設定します	CALL TRANSPOSE ..50		






# CALL AUDREG

Audio Register(オーディオレジスタ:音楽LSI用の登録)

- 
- 働き**  音楽LSIのレジスタに書き込みを行います。
- 書き方**  CALL AUDREG (<レジスタ番号>, <値> [, <チャンネル番号>])
- 例** CALL AUDREG (&H20, 0)
- 説明**  FM音源で演奏する音楽LSIのレジスタに対して書き込みを行います。
- システムソフトウェアが割り込みなどでひんばんに書き込んでいるレジスタには、効果がないときがあり、このときはシステムの再立ち上げが必要な場合があります。
- <チャンネル番号>は、0または省略しなければなりません。これは、MSX-AUDIOとの互換性をとるために用意されているもので、MSX-MUSICでは意味を持ちません。
- レジスタへの書き込みは、音楽LSIについての知識を得てから行ってください。

# CALL BGM

BGM(ビジー-エム:Back Ground Music の略)

- 働き  バックグラウンド処理を行うかどうかを指定します。
- 書き方  CALL BGM(<変数>)
- 例 CALL BGM(0)
- 説明  <変数>は0または1の値をとり、次のような意味を持ちます。
- 0 : PLAY文のバックグラウンド処理を行わない。
  - 1 : PLAY文のバックグラウンド処理を行う。
- CALL MUSIC命令による初期化ではバックグラウンド処理(1)が指定されています。
- バックグラウンド処理とは、PLAY命令により演奏を開始した後、演奏が終了する前に他の命令を実行することができる方法です。

## サンプルプログラム

40ページのプログラムを変更して、演奏するデータを表示させます。



CALL BGM(1)を0に変えると、演奏が途切れます。これは、演奏を終了してからデータを表示し、その後、演奏を再開するためです。

```
100 CALL MUSIC(0,0,1,1,1)
105 CALL BGM(0)
110 CALL TRANSPOSE(2400)
120 PLAY #2,"T90","T90","T90"
130 READ A$:IF A$="END" THEN RESTORE:GOT
0 130
140 READ B$,C$
150 PLAY #2,A$,B$,C$
155 PRINT A$,B$,C$
160 GOTO 130
170 DATA L16,L16,L16
180 DATA 05C8DC04B05CA4F8
190 DATA 04F4F805C404A8
200 DATA 03A8B-AG+A04F4C8
210 DATA 05E4D8L4G.L16
220 DATA A4G8L4B-.L16
230 DATA 03B-4B-804L4D.L16
240 DATA L804B-05C.D16EF.G16
250 DATA L804DF.F16GB-.B-16
260 DATA L803GA.B-1604CD.E16
270 DATA END
```



# CALL MUSIC

Music(ミュージック:音楽)

働き   
書き方 

MSX-MUSICのシステムを初期化します。


CALL MUSIC([(<モード>)[, [0]][, <PLAY文第1文字列のチャンネル数>[, <PLAY文第2文字列のチャンネル数>[, ...[, <PLAY文第9文字列のチャンネル数>]]]]]]]]))

例

CALL MUSIC 初期化します。

CALL MUSIC(0,0,1,1,1,1,1,1,1,1)

1チャンネルずつPLAY文の文字列に割り当てます。

説明 

音楽LSIの初期化とともに9個のFM音源のチャンネルをどのように使用するかを指定します。CALL MUSIC命令により初期化を行うまでは、全てのMSX-MUSICの命令は使用できません。

<モード>は0か1で、0はリズム音を使用しないモードです。

1はリズム音を使用するモードです。

リズム音を使用する場合にはチャンネル7、8、9を使用しますので、通常の和音に使えるのは残り6チャンネルになります。したがって、PLAY文で指定できる文字列はリズム使用時には6以下、リズムを使わないときは9以下である必要があります。

[0]は、MSX-AUDIOとの互換性のため、必ず0です。

チャンネルの使用割り当ては、PLAY文では、チャンネル番号の小さい方から(1,2,3...を)割り当てます。

悪い例)

PLAY文の文字列へのチャンネル数を、0に設定したり省略したりすることはできません。次の例を参照してください。

CALL MUSIC(0,0,0,5,0)

↑ ↑

0は設定できません

(Illegal function callになります)

CALL MUSIC(0,0,1,,2)

↑

省略できません(Syntax errorになります)


## CALL MUSIC

パラメータなしでCALL MUSICと使われた時は、CALL MUSIC

(1, 0, 1, 1, 1)と同じになります。

つまり、

(1, 0, 1, 1, 1)



- FM音源のチャンネル1をPLAY文の最初の文字列に割り当てる。
- FM音源のチャンネル2をPLAY文の2番目の文字列に割り当てる。
- FM音源のチャンネル3をPLAY文の3番目の文字列に割り当てる。
- リズム音をPLAY文の4番目の文字列に割り当て使用する。
- PLAY文の5番目～7番目の文字列はPSG音源の制御に割り当てる。

という意味になります。

### 注意！




CALL MUSIC命令を実行すると、音楽演奏用のワークエリア(約810バイト)を確保します。


このため、それ以前に使われていた変数、配列などの内容およびFOR～NEXT、GOSUBの情報がクリアされます。CALL MUSIC命令は、プログラムの初めに実行してください。



# CALL PITCH

Pitch(ピッチ:調子、音の高低)

- 働き  FM音源の音の高さ(ピッチ)を与えます。
- 書き方  CALL PITCH(<ピッチ1>[,<ピッチ2>])
- 例 CALL PITCH(440)
- 説明  FM音源で発生する全体の音の高さを指定します。<ピッチ1>の範囲は410~459で単位は[Hz]です。中央C(ハ長調)のすぐ上のA音(a2)の周波数を基準に音の高さを表します。トランスポーズ(移調)とは独立に設定でき、初期値は440です。ピッチを変えるとFM音源の音の高さが変化します(リズム音や音程をもたない音は除く)。また、この命令は、PSG音源の音には作用しません。
- <ピッチ2>はMSX-AUDIOとの互換性のために用意されているもので、MSX-MUSICでは指定しても意味を持ちません。
- トランスポーズについては、CALL TRANSPOSE文の項を参照してください。




参照  CALL TRANSPOSE (50ページ)

注意 CALL PITCH命令を実行すると、実行した次点で演奏中のピッチが変化します。つまり、PLAY命令の実行により演奏データが一旦、バッファに貯えられ、順次、演奏されますが、演奏終了前にCALL PITCH命令が実行されると、それ以降のデータは変更されたピッチで演奏されます(次のプログラムの140行を止めると良く判ります)。

```
100 CALL MUSIC
120 CALL PITCH(410)
130 PLAY #2,"cc"
140 CALL PLAY(0,A):IF A<>0 THEN 140
150 CALL PITCH(459)
160 PLAY #2,"cc"
```

# CALL PLAY





PLAY(プレイ:演奏している)

- 
- 働き**  PLAY文による音楽を演奏中かどうかを返します。
- 書き方**  CALL PLAY(<PLAY文の文字列番号>, <変数名>)
- 例** CALL PLAY(0, A):PRINT A
- 説明**  FM音源およびPSGの状態を調べ、指定されたチャンネルが音楽を演奏中かどうかを判断します。演奏中であれば-1、そうでなければ0の値を変数に代入します。<文字列番号>は、PLAY文で演奏を指定したデータ(文字列)の番号です。(例、PLAY #2, "AB", "CD", "EF" とすると"AB"が「1」、"CD"が「2」、"EF"が「3」です。) <文字列番号>として0が与えられた場合は、FM音源およびPSG音源が演奏中であれば-1を、そうでなければ0を返します。

CALL PLAY文の文字列番号は、MUSIC文で指定したチャンネル数+3まで使えます。つまり、MUSIC文で指定したFM音源に加え3チャンネルのPSG音源について有効です。

# CALL STOPM


Stop Music(ストップ ミュージック:音楽を止める)

- 
- 働き**  バックグラウンドで実行中のPLAY文の演奏を停止します。
- 書き方**  CALL STOPM
- 説明**  バックグラウンドで実行中のPLAY文の音楽の演奏を停止します。
- 参照**  CALL BGM (44ページ)




# CALL TEMPER

Temperament (テンペラメント: 音律)

働き 


音律を設定します。

書き方 

CALL TEMPER (<音律番号>) 音律番号: 0 ~ 21

例

CALL TEMPER (0)

説明 

音律を与えるステートメントで、FM音源の音の高さに影響を与えます。





音律は1オクターブをどのような比率で12音に分割するかを決めるもので、古典音楽には古典音律が適していると言われます。

初期値は9番の完全平均律です。

番号	音 律		
0	ピタゴラス		
1	ミーントーン		
2	ヴェルクマイスター		
3	ヴェルクマイスター(修正)		
4	ヴェルクマイスター(別)		
5	キルンベルガー		
6	キルンベルガー(修正)		
7	ヴァロッティ・ヤング		
8	ラモー		
9	完全平均律(初期値)		
10	純正律 c	メジャー	(aマイナー)
11	純正律 c i s	メジャー	(bマイナー)
12	純正律 d	メジャー	(hマイナー)
13	純正律 e s	メジャー	(cマイナー)
14	純正律 e	メジャー	(c i sマイナー)
15	純正律 f	メジャー	(dマイナー)
16	純正律 f i s	メジャー	(e sマイナー)
17	純正律 g	メジャー	(eマイナー)
18	純正律 g i s	メジャー	(fマイナー)
19	純正律 a	メジャー	(f i sマイナー)
20	純正律 b	メジャー	(gマイナー)
21	純正律 h	メジャー	(g i sマイナー)

# CALL TRANSPOSE


Transpose(トランスポーズ:置きかえる。移調する)

- 働き  FM音源の音に対してセント単位で移調を与えます。
- 書き方  CALL TRANSPOSE(<トランスポーズ値1>[,<トランスポーズ値2>])
- 例 CALL TRANSPOSE(0)
- 説明  セント(半音の1/100)単位で、移調を行います。  
たとえば、1オクターブ上げるには、+1200を与えます。  
<トランスポーズ値>は、-12799~+12799です。ただし、音色によってはある高さの範囲以外は制限されることがあります。  
トランスポーズはピッチとは独立して設定できます。  
初期値は0です。  
<トランスポーズ値2>はMSX-AUDIOとの互換性をとるために用意されているもので、指定しても無視されます。
- 参照  CALL PITCH (47ページ)




# CALL VOICE

Voice(ボイス:声、音)

働き 


FM音源の各チャンネルに音色を直接設定します。

書き方 

CALL VOICE ([<チャンネル1用ボイスの数式>], [<チャンネル2用ボイスの数式>], ..., [<チャンネル9用ボイスの数式>])

例

CALL VOICE (@0, @0, @0, ..., @7, @7, @7)

説明 

音楽用LSIの9チャンネルあるFM音源のそれぞれに音色を設定します。

音色の設定方法には2つあります。

あらかじめ設定されている音色を使って設定する場合、0~63の音色番号(次ページ参照)を、数式または、変数名により指定します。数式で表すときは、式の前に@記号をつけて、配列変数名と区別します。

例. CALL VOICE (@14, @14, @14)

プログラムにより音色データ(55ページ参照)を与えて設定する場合には、配列変数に音色データを入れて、その配列変数名を指定します。音色データの送り方は、CALL VOICE COPY文を参照してください。

パラメータを省略したチャンネルの音色は変更されません。

音色番号のうち\*印が付いていない音色や、配列変数で設定する場合は同時に1音色しか設定できません。複数設定したときは、パラメータ列のいちばん右側のパラメータか、最後に実行したCALL VOICE文の設定のみが有効となります。

例. • CALL VOICE (@26, @27)

チャンネル1の音色は2と同じ27番の音色になります

• CALL VOICE (@20, @10)

CALL VOICE (., @21)

上記のように続けて命令すると、チャンネル1の音色は3と同じ21番となります。

• CALL VOICE (A., B)

チャンネル1の音色は3と同じBという配列変数で設定される音色となります。

# CALL VOICE

## 音色一覧表

(VOICE文およびMML(ミュージック・マクロ・ランゲージ、57ページ参照)の@nで使用できる音色の一覧表です。)

音色番号	音色名
0 *	ピアノ 1
1	ピアノ 2
2 *	バイオリン
3 *	フルート 1
4 *	クラリネット
5 *	オーボエ
6 *	トランペット
7	パイプオルガン 1
8	シロフォン
9 *	オルガン
10 *	ギター
11	サンツール 1
12 *	エレキベース
13	クラビコード 1
14 *	ハーブシコード 1
15	ハーブシコード 2
16 *	ピブラフォン
17	琴 1
18	太鼓
19	エンジン 1
20	UFO
21	シンセサイザベル
22	チャイム
23 *	シンセ・ベース
24 *	シンセサイザー
25	シンセ・ドラム
26	シンセ・リズム
27	ハーモ・ドラム
28	カウベル
29	ハイハット
30	スネア・ドラム
31	バス・ドラム

音色番号	音色名
32	ピアノ 3
33 *	ウッドベース
34	サンツール 2
35	ブラス
36	フルート 2
37	クラビコード 2
38	クラビコード 3
39	琴 2
40	パイプオルガン 2
41	P o h d s P L A
42	P o h d s P R A
43	チャーチオルガン L
44	チャーチオルガン R
45	シンセ・バイオリン
46	シンセ・オルガン
47	シンセ・ブラス
48 *	ホルン
49	三味線
50	マジカル
51	フワフ
52	ワンダーフラット
53	ハードロック
54	マシーン
55	マシーン V
56	コミック
57	SE-コミック
58	SE-レーザー
59	SE-ノイズ
60	SE-星 1
61	SE-星 2
62	エンジン 2
63	無音


//////  
※音色名は参考のために付けたもので音色によっては、実際の楽器の音と異なることがあります。


\*印はFM音源が内蔵している音色で、\*印のないものはプログラムが自動的に合成し、作り出す音色です。



# CALL VOICE COPY

Voice Copy (ボイスコピー: 音色を転送する)


働き  音色データの内容をコピーします。

書き方  CALL VOICE COPY (<パラメータ 1><パラメータ 2>)

例 CALL VOICE COPY (@ 17, @ 63)  
17番の音色データを63番に転送する。

DIM A%(16): CALL VOICE COPY (@ 28, A%)  
28番の音色データを配列変数A%に転送する。

CALL VOICE COPY (A%, @ 63)  
配列変数A%の音色データを63番に転送する。

説明  配列と音色番号(0~63)の間で、音色データの転送を行います。  
<パラメータ 1>の音色データを<パラメータ 2>に転送します。@  
と数式が指定された時は、その数式の結果で指定される音色番号  
の音色データが対象となります。

<パラメータ 1>に指定できる音色番号は0~63のうち\*印の付  
いていない音色の番号です。<パラメータ 1>に\*印の付いている  
音色番号を指定すると "Illegal function call" となります。

<パラメータ 2>に指定できる音色番号は63に限ります。

@記号がない場合の変数名は配列変数とみなされ、その内容が転  
送の対象になります。

1つの音色データは32バイトの長さがあります。

## サンプルプログラム


```
100 CALL MUSIC
110 CLEAR: DIM A%(15)
120 CALL VOICE COPY (@25, A%)
130 A%(8) = 2200
140 CALL VOICE COPY (A%, @63)
150 PLAY #2, "@25cdef"
160 PLAY #2, "@63cdef"
```






# PLAY \*

Play(プレイ:演奏する)

働き 


音楽をMMLにしたがって演奏します。

書き方 

PLAY [#<モード>] <文字列1> [, <文字列2>] [, <文字列3>] · [, <文字列13>]

例

PLAY #2, "CD", "EF", "GA"

説明 

PLAY文は音楽を演奏するもので、FM音源9音、従来のPSG音源3音の最大12音まで同時演奏が可能です。

<文字列>に書かれたミュージック・マクロ・ランゲージ(MML)にしたがって演奏します。

他の拡張命令と異なりCALL文は必要ありません。

<モード>は0~3の値で、PLAY文の音源や動作モードを次のように設定します。

- 0 (または省略時)はPSGのみが音源となり、文字列は最大3つまでとなります。MSX、MSX2のPLAY文と互換性があります。
- 1の時、"Illegal function call"となります。これはMSX-AUDIOとの互換性をとるために用意されているものでMSX-MUSICではエラーとなります。
- 2又は3の時、FM音源、リズム音、PSG音源を使用できます(2の時と3の時で動作に違いはありません)。

<文字列>と音源との関係は初めから順に、

<FM音源用文字列1>, ..., <FM音源用文字列n>, <リズム音用文字列>, <PSG音源用文字列1>, <PSG音源用文字列2>, <PSG音源用文字列3>

となります。nはCALL MUSIC文で設定されたチャンネル数です。

CALL MUSIC文でリズム音を使用しないモードに設定した場合は、リズム音用文字列をカンマと共に省略しなければいけません。

例としてパラメータ省略時のCALL MUSIC文に対する文字列の配列をあげると次のようになります。

<FM音源用文字列1>, <FM音源用文字列2>, <FM音源用文字列3>, <リズム音用文字列>, <PSG音源用文字列1>, <PSG音源用文字列2>, <PSG音源用文字列3>



## PLAY文で使用できるMML(ミュージックマクロランゲージ)の一覧表

文字	意味	値の範囲	初期値*1
M n	エンベロープ周期の設定*2	$1 \leq n \leq 65535$	M 255
S n	エンベロープ形状の設定*2	$0 \leq n \leq 15$	S 0
V n	音量の設定	$0 \leq n \leq 15$	V 8
L n	長さの設定	$1 \leq n \leq 64$	L 4
Q n	音の長さの割合	$1 \leq n \leq 8$	Q 8
O n	オクターブの設定	$1 \leq n \leq 8$	O 4
>	オクターブを1つ上げる		
<	オクターブを1つ下げる		
T n	テンポの設定	$32 \leq n \leq 255$	T 120
N n	nで指定された高さの音を発生する	$0 \leq n \leq 96$	
R n	休符の設定	$1 \leq n \leq 64$	R 4
A ~ G	音程の発生		
+, #	音を半音上げる		
-	音を半音下げる		
.	音符や休符の長さを1.5倍にする (ピリオド)		
X A \$ ;	文字変数 A \$ に入っている M M L を演奏する*3		
= x ;	パラメータ n を変数 x で設定する	*4	
&	タイ、前後の音をつなぐ		
{ } n	連符、n分音符を { } 中の音程の個数で等分にした音を発生する	$1 \leq n \leq 64$	L n で設定された値
@ n	n番の音色に切り替える	$0 \leq n \leq 63$	
@ V n	音量を細かく設定する	$0 \leq n \leq 127$	
@ W n	nで指定された長さだけ状態を維持する	$1 \leq n \leq 64$	L n で設定された値
Y r, d	音楽用 L S I のレジスタ r に d を書き込む		

\*1 : 最初に設定されている値(CALL MUSICを使ったときに設定される値)。

\*2 : P S G 音源専用です。

\*3 : 「X A \$ ;」の後ろにMMLを続けるとエラーになります。

\*4 : 値のとり範囲は直前のMMLにより決定されるが、32767を越えることはできません。

# PLAY

## リズム音用MML一覧

リズム音の場合、5種類のリズム楽器の音のうち3種類までを同時に発生できます。このため、リズム音では、まず同時に鳴らしたい楽器名を並べ、その後に次の音を発生するまでの(または終了するまでの)待ち時間を続けます。


文字	意味	値のとり範囲	初期値
B	バスドラム音を発生		
S	スネアドラム音を発生		
M	タムタム音を発生		
C	シンバル音を発生		
H	ハイハット音を発生		
n (数値)	直前までに書かれた音楽を発生し、n分音符分待つ	$1 \leq n \leq 64$	
V n	音量を設定する	$0 \leq n \leq 15$	8
!	直前に書かれた音楽の音量をアクセントボリュームにする		
@ A n	アクセントのついている音楽の音量を設定する	$0 \leq n \leq 15$	

「T n」、 「@ V n」、 「R n」、 「X A \$ ;」、 「= x ;」、 「. (ピリオド)」は前ページのMMLと同じです。

例: PLAY # 2, "", "", "", "B S H 8 H 8 S ! H ! 8 H 8"

- バス、スネア、ハイハットを鳴らし、8分音符分待ちます。
- ハイハットを鳴らし、8分音符分待ちます。
- スネア、ハイハットをアクセントつきで鳴らし、8分音符分待ちます。
- ハイハットを鳴らし、8分音符分待ちます。

# BASICコンパイラの使い方



この「BASICコンパイラの使い方」では、MSX BASICで作ったプログラムを高速で実行するBASICコンパイラについて説明します。BASICコンパイラは、パソコンのメモリ上にあるBASICプログラムをその場で機械語プログラムに変換して実行します。BASICでリアルタイムのゲームやコンピュータグラフィックスのプログラムを作ることができます。

コンパイラとは? -コンパイラの特長-	60
コンパイラの使い方	62
●コンパイラの使い方	62
●コンパイラ用の命令	63
●使える命令と使えない命令	66
●こんなこともできます	68
●プログラム作成時のご注意	72



# コンパイラとは？

—コンパイラの特長—

パソコンのプログラム言語として良く知られているものにBASICがありますが、この他にFORTRAN、COBOL、C言語などもあります。


しかし、実際にはパソコンはこれらのプログラム言語を理解できる訳ではなく、機械語というパソコン用の言語しか判りません。そこで人の作ったBASICなどのプログラムを機械語に変換してあげる必要があります。

BASICとこれらのプログラム言語の大きな違いの一つに、プログラムの変換(実行)の仕方があります。BASICではプログラム命令を一つ一つ実行するインタプリタ方式で動作します。これに対して、FORTRANなどはプログラム命令をいったん機械語に変換して、いっきに命令を実行するコンパイラ方式で動作します。

インタプリタ方式の場合は命令を一つ一つ実行するので、プログラム作成が途中でエラーが発生するところまでの結果を見ることができます。しかし、命令を実行するときの一つずつBASIC命令から機械語に変換するので、実行速度が遅くなります。


コンパイラ方式の場合はプログラムを全て機械語に変換して、変換したプログラム(機械語)で実行するので、実行速度が高速になります。しかし、プログラムがある程度完成していないとエラーなどで実行できず、途中までの結果も見ることができません。

## ● インタプリタの場合




## ● コンパイラの場合

### ① 機械語に変換



### ② 命令の実行



## ちょっと一言 ■ コンパイラとインタプリタ

コンパイラ(Compiler)は翻訳という意味で、インタプリタ(Interpreter)は通訳という意味です。

つまり、プログラムを機械語に翻訳(全部を変換)して実行するか、通訳(一つずつ変換)して実行するかの違いがあります。

## BASICコンパイラの特長

BASICコンパイラは、メモリ上のBASICプログラムをその場で機械語プログラムに変換し、機械語のプログラムとして高速で実行します。入出力命令などの一部のものを除き、普通のMSX BASICで書かれたプログラムをほとんど変更することなく機械語に変換できます。

使用方法も簡単で、従来のコンパイラ(変換ソフト)のように、

- ①ソースプログラム(BASICプログラム)の作成
- ②コンパイル→オブジェクトファイル(機械語プログラムに変換)の作成
- ③オブジェクトファイルの実行

といったステップは不要です。

プログラムの実行速度がBASICの15~20倍となり、リアルタイムのゲーム、膨大な計算を必要とするコンピュータ・グラフィックス、デモプログラムなどに使用できます。

特長としては、

- 実数および文字変数が使用できます。
- 割り込みが使用できます。
- 配列の次元は自由で、メモリ容量によってのみ制限されます。
- MSX BASICと基本的に文法が同じため、MSX BASICでプログラムの作成・修正をして、BASICコンパイラで実行させることもできます。
- プログラムの一部だけをコンパイルして実行させることもできます。

BASICコンパイラがサポートしていない命令・関数はMSX BASICで実行して、高速化が必要な部分をコンパイルして実行することができます。

制限としては、

- 実数の精度が低い(有効数字約4桁。2進法で言えば4.5桁)。
- ファイル入出力関係の命令・関数がサポートされていません。
- コンパイルした機械語プログラムのSAVE、LOADができません。
- MSX BASICと少し文法の違う命令があります。

[ご注意] すでに作られているBASICプログラムの全てがコンパイルして実行できる訳ではないので注意してください。お使いになる前に、この章の説明・注意をよく読んでください。

ちょっと  
一言

### ■ソースとオブジェクト

いきなりカタカナ用語が出ていますが、パソコン用の用語なのでこのまま覚えてください。ソース(Source:元の)は変換する前のプログラムの意味です。オブジェクト(Object:目的の)は実行するためのプログラムの意味です。



# コンパイラの使い方

パソコンの電源を“入”にした後、キーボードから CALL BC 』と入力します。画面に“Ok”と表示されたら、パソコンに内蔵されている BASIC コンパイラが使用可能となります (CALL BC はプログラム中でも実行できます)。

## コンパイラの使い方

CALL BC 命令を実行したら、次のプログラムを実行してコンパイラの速さ確かめてみましょう。

```
1 0  TIME = 0
2 0  DEFINT A-Z
3 0  DEFBNEXT A TO 250
4 0  B=NEXT A
5 0  PRINT B, TIME
6 0  END
```

CALL RUN 』とキー入力するとプログラムがコンパイル・実行され、0 から 250 の合計と計算に掛かった時間 (1/60秒の単位) が表示されます。普通に RUN 』で実行したときとの差を見てください。

CALL RUN 』とキー入力したとき、BASIC コンパイラで機械語に変換 (コンパイル) できない命令やプログラムにエラーがあれば、変換中または実行中にエラーメッセージを出して実行を中断します。

次のプログラムでは PLAY 命令がコンパイルできないので、プログラムを一部だけコンパイルする TURBO ON 命令を使います。コンパイルできない命令を TURBO ON 命令の外側に入れておきます。

```
1 0  SCREEN 5
2 0  CALL TURBO ON
3 0  CIRCLE (126, 106), C*10, C
4 0  C=C+1
5 0  IF INKEY$ <> "" THEN END
6 0  IF C < 15 THEN 30
7 0  CALL TURBO OFF
8 0  PLAY "CDEFG": GOTO 10
9 0  END
```

このプログラムを RUN 』すると 20 ~ 70 行の間だけがコンパイルされ、PLAY 命令のみ普通の BASIC と同様に実行されます。一つのプログラムの中に TURBO ON/OFF の組は、いくつできててもかまいませんが、プログラムを CALL RUN することはできなくなります。

RUN 』または RUN "ファイル名" 』で実行できます。




## コンパイラ用の命令


BASICコンパイラ用の命令は、次の4つだけです。(拡張命令については、68ページを参照してください。) また、CALLは"\_(アンダーバー)"でも代用できます。

### CALL BC

BC (BASICコンパイラ)

働き  BASICコンパイラを使用可能にします。

書き方  CALL BC


説明  この命令により、BASICコンパイラが使用可能な状態となります。一度実行すると、リセットしたり電源を切らない限り、BASICコンパイラの機能が使用できます。この命令は、BASICコンパイラをシステムに登録するのみです。メモリの確保はしません。BASICコンパイラは基本的には、MSX BASICのみをサポートしています。MSX2、MSX2+特有の命令は使用できません。ただし、SCREEN命令に関しては12まで指定できます。また、漢字モードの状態でもBASICコンパイラの使用はできますが、逆(BASICコンパイラでCALL KANJI)はできません。


MSX-DOSおよびRAMディスク機能、メモリ・マッパー機能とは同時に使えません。


MSX-DOSなどの機能を終了(解除)したときに、CALL BC命令の再実行が必要です。

### CALL RUN

RUN (ラン:実行)


働き  BASICプログラムをすべてコンパイル(機械語に変換)して、プログラムを実行します。


書き方  CALL RUN


説明  パソコンのメモリ上にあるBASICプログラムを、全て機械語に変換し、実行します。この命令を実行する前に、CALL BC命令を実行しておく必要があります。また、この命令はプログラム中で実行することはできません。

# CALL TURBO ON CALL TURBO OFF

TURBO ON/OFF (ターボ・オン/オフ)

働き  BASICプログラム中で、プログラムの一部をコンパイルします。

書き方  CALL TURBO ON [( <変数> ) ( , <変数> ) … ( , <配列> ) ]  
CALL TURBO OFF

説明  プログラム中にBASICコンパイラでコンパイルできない命令や関数があるとき、プログラムの一部をコンパイルして機械語として実行します。

TURBO ONがコンパイルの開始を、TURBO OFFがコンパイルの終わりを示します。

CALL TURBO ON/OFFはプログラム中でのみ使用でき、ダイレクトモードでは実行できません。

CALL TURBO ON、CALL TURBO OFF命令はマルチステートメント中には置けません。必ず独立した1行の中におさめてください。

悪い例) 1 0 0 IF A = 0 THEN CALL TURBO  
OFF

良い例) 1 0 0 IF A = 0 THEN 1 1 0 ELSE ……  
1 1 0 CALL TURBO OFF

CALL TURBO ON、CALL TURBO OFF命令は文章中に何個でも指定できますが、入れ子の状態にしては行けません。また、GOTO命令などでTURBO命令で囲まれた中から、範囲の外へ実行を移すことはできません。

悪い例)

```
1 0 SCREEN 5
2 0 CALL TURBO ON
3 0 CIRCLE (126, 106), C*10, C
4 0 C=C+1
5 0 IF INKEY$ <> "" THEN 90
6 0 IF C < 15 THEN 30
7 0 CALL TURBO OFF
8 0 PLAY "CDEFG": GOTO 10
9 0 END
```

上の例では50行のTHEN 90により、TURBO OFF命令のある70行でUndefined line number エラーとなります。50行の90をENDに修正すると、動作します。



TURBO ON/OFF命令を使うとき、変数や配列の使い方に次の制限があります。

普通のBASICの部分と、TURBO命令で囲まれた部分では、変数や配列は別々に定義されます。同じ変数名で指定しても異なった値を持ちます。

また、TURBO OFFで1つのブロックが終わると、その中で使われていた変数や配列はすべて無効となります。

ただし整数型の変数と配列に限って、共通の変数として宣言することができます。このときはCALL TURBO ONの後に必要な変数、配列名をカッコでくくって並べます。

ただし、TURBO命令の前に宣言文などの実行文で使用された変数、配列のみです。

例) CALL TURBO ON (A,B%,C(),DE%())

BASICコンパイラで扱われる実数型の変数は普通のBASICとは少し違っています。

型は"! "マークで表す単精度の変数も"# "の倍精度の変数も、精度は約4桁で、範囲は $2.939E-39 \sim 1.70E+38$ です。

また、10,000以上の数は「指数形式」で表示されます。精度が多少低いので、普通のBASICとは違った計算結果となることもあります。

配列の宣言を省略することはできません。

配列はプログラムの一番初めの行(CALL TURBO ONのときはその直後)に宣言してください。宣言していない配列は扱うことができません。配列は何次元でもメモリの許す限り使えます。

無限ループを作る(10 GOTO 10のような)と、**CTRL** + **STOP**でもプログラムを止めることができなくなります。必要な場合はキー入力を受けつける部分を作ってください。

例) 10 IF INKEY\$ = " " THEN 10

## 使える命令と使えない命令

BASICコンパイラは、基本的にMSX BASICの命令と関数をサポートしていますが、コンパイルできない(機械語に変換できない)命令、関数があります。次の命令を含んだ状態でコンパイルするとエラーとなります。

### 使えない命令

#### A

AUTO

#### B

BASE、BLOAD、BSAVE

#### C

CALL、CDBL、CINT、  
CLEAR、CLOAD、CLOAD?、  
CLOSE、CONT、CSAVE、  
CSNG、CVD、CVI、CVS

#### D

DEFFN、DELETE、DRAW、  
DSKF

#### E

EOF、EQV、ERASE、ERL、  
ERR、ERROR

#### F

FIELD、FILES、FPOS、  
FRE

#### G

GET

#### I

IMP、INPUT#

#### K

KEYLIST、KILL

#### L

LFILLES、LINE INPUT#、  
LIST、LLIST、LOAD、LOC  
LOF、LPRINT USING、  
LSET

#### M

MAXFILES、MERGE、  
MKD\$、MKI\$、MKS\$、  
MOTOR

#### N

NEW、NAME

#### O

ON ERROR GOTO、OPEN

#### P

PLAY、PRINT#、PRINT#  
USING、PRINT USING、  
PUT KANJI

#### R

RENUM、RESUME、RSET

#### S

SAVE、SPC、

#### T

TAB、TRON、TROFF

#### W

WIDTH



## 使い方が変わる命令

つぎの命令、関数は使用方法が限られます。

CIRCLE	開始、終了の角度および比率は指定できません。
COPY	グラフィック画面上でのCOPY命令以外は使えません。 (ディスクや配列へのコピーはできません。)
DEFDBL	DEFSNGと同じです。A!とA#は同じ変数を意味します。 (計算を速く行いたいときは、DEFINT命令で整数宣言した変数を使ってください。)
DIM	すべての実行文に先だって宣言します。 添字は整数の定数(変数でない)で指定します。 DIM文の前に置けるのはDEFINT、DEFSNG、DEFDBL、DATA、DIM、コメント文だけです。
INPUT	一度に代入できる変数は1つだけです。
KEY	ON KEY GOSUB、KEY(N) ONなど以外では使えません(例、ファンクションキーに命令は設定できません)。
LOCATE	X、Y座標は必ずセットで指定し、カーソルスイッチは指定できません。
NEXT	NEXTの後の変数名は省略できません。
ON	ON STOP GOSUB命令は使えません。
PRINT	カンマの働きが違います。普通のBASICではカンマは有効数字の最大桁数に相当するスペースを出力しますが、BASICコンパイラではタブコードを出力するだけです。 また、数値を出力するとき画面上のラップは行われません。
PUT	PUT SPRITE命令以外は使えません。
RUN	変数の初期化は行われません。プログラム中にあると、一番初めの行に実行が移りますが、変数の初期化はしません。
SCREEN	画面モード、スプライトサイズ以外は指定できません。画面モードは、MSX2+のSCREEN 12まで指定できます。
SET	SET PAGE命令とSET SCROLL命令以外は使えません。
STOP	END文と同じ意味になります。
USR	引数として使える値は整数のみです。
VRPTR	ファイル番号を引数として指定することはできません。

これ以外の命令、関数は普通のBASICと同じように扱うことができます。

## こんなこともできます


BASICコンパイラにはCALL BCとCALL RUN、CALL TURBO ON/OFFという拡張命令がありますが、このほかに次の拡張命令があります。


- Y座標のクリッピングをON/OFFする ..... #C
- NEXT文のオーバーフローをチェックする ..... #N
- プログラム中に機械語プログラムを入れる ..... #I

BASICコンパイラの拡張命令は、コメント文を「#」で始めることで指定されます。(コメント文は「REM」で始めても「'」でもかまいません。) また、「#」の後にくるキャラクタは大文字でも小文字でもかまいません。


### #C

Clip (クリップ:止める)

働き  Y座標指定のクリッピングを行うかどうかを指定します。

書き方  ' #C + ..... クリッピングON (MSX BASICと同じ)  
' #C - ..... クリッピングOFF

例 100 ' #C +

説明  クリッピングとは、LINE命令などを実行するとき、Y座標の指定値が表示範囲を越えるときに、値を制限して命令を行う機能です。

「#C+」でクリッピングをONにし(MSX BASICと同じ)、  
「#C-」でクリッピングをOFFにします。BASICコンパイラの起動時にはONが設定されています。

例).

```
10 SCREEN 8
20 ' #C -
30 LINE (0,0) - (255,255)
40 IF INKEY$="" THEN 40
50 ' #C +
60 LINE (0,0) - (255,255)
70 IF INKEY$="" THEN 70
```


### 注意


BASICコンパイラでは、SCREEN 0~3の画面でのグラフィック処理、PAINT命令、CIRCLE命令の実行には、MSX BASICと同じ処理を行うためクリッピングの指定は意味を持ちません。




## # N

Next (ネクスト:次の)

働き  NEXT文の実行時に変数のオーバーフローをチェックします。

書き方  '#N+ ..... オーバーフローをチェックします。  
'#N- ..... オーバーフローをチェックしません。

例 1 0 0 '#N+

説明  FOR~NEXT文のコンパイル時にオーバーフローをチェックするコードを作成するかどうかを決めます。

「#N+」とすると、それ以降のNEXT文のコンパイルにおいてオーバーフローをチェックし、オーバーフローならループから脱出するプログラムを作成します。

「#N-」とすると、オーバーフローをチェックしないプログラムを作成します。BASICコンパイラ起動時には「#N+」が設定されています。

例).

```
1 0 FOR I%=0 TO &H7FFF
2 0 NEXT I%
```

このプログラムをMSX BASICで実行させると、ループの終了直前にエラーが起きます。これは、&H7FFFに1をたした結果、値が整数の範囲を越えたことによって起こります。

BASICコンパイラで実行させると、無限ループになります。&H7FFFより大きい整数はコンパイラでは存在しないからです。


このような状態を回避したい時にこの命令を用いてください。


注意 「#N+」とすると、コンパイルするプログラムのサイズ、実行時間共に増加します。

また、この命令を使ったプログラムはMSX BASICと異なった動作となります。


## 井 I

Inline(インライン:文中に入る)

働き  プログラム中に機械語プログラムを記述します。

書き方  '# I ..... この行に機械語のプログラムを記述します。

例 100 '# I &H1B

説明  数字を書くとその値(バイト)がメモリに置かれます。変数名(整数型のみ)を書くと、コンパイル時にその変数の数値格納アドレスがプログラムに置かれます。

"@" が頭に付いた10進数を書くと、その数で示されるBASICSプログラムの行番号に実行が移ることを意味します。この行番号はコメント行にあるため、RENUM命令では行番号の変更がされないので注意してください。

例).グラフィック画面用文字出力ルーチン

```
10 SCREEN 5 : X% = 0 : Y% = 0
20 PRESET (0,0)
30 FOR C% = 0 TO 255
40 GOSUB 100
50 NEXT C%
60 IF INKEY$ = " " THEN 40
70 END
100 '# I &H2A, C%, &H29, &H29, &H29
, &H11, &HBF, &H1B, &H19
110 '# I &H06, &H08, &H3A, &HB7, &
HFC, &H32, X%, &H3A, &HB9, &HFC, &H
32, Y%
120 '# I &H7E, &H23, &HC5, &H06, &
H08
130 '# I &H17, &HF5, &HC5, &HE5, &
HDC, @170, &HE1, &HC1, &H3A, X%, &H
3C, &H32, X%, &HF1
140 '# I &H10, &HED, &H3A, X%, &HD
6, &H08, &H32, X%, &H3A, Y%, &H3C, &
H32, Y%, &HC1, &H10, &HD6
150 '# I &H3A, X%, &HC6, &H08, &H3
2, &HB7, &HFC, &H3A, Y%, &H38, 2, &H
D6, &H08
160 '# I &H32, &HB9, &HFC, &HC9
170 PSET (X%, Y%) : RETURN
```

© ASCII Corp.



---

前ページのプログラムは、BASICコンパイラを使うとき、OPEN命令、PRINT#命令が使えないので、それを補うためのグラフィック画面に文字を出力するサブルーチン用プログラムです。

20行のPRESET文で出力位置を決め、30～50行のFOR～NEXTループでC%に文字のキャラクタコードを代入しています。次に40行で機械語サブルーチンを呼び出しています。

100～160行までが機械語サブルーチンですが、その中からさらに170行のBASICサブルーチンを呼び出す形式になっているので、自作のプログラムに組み込む際には170行を忘れないでください。

この例の場合には、130行中の"@170"です。この"@170"はRENUM命令で修正されないので注意してください。

#### 注意点

- X%、Y%は整数型であれば別の変数名でもかまいません。
- このプログラムでは、X座標は0～255までです。
- 重なったグラフィック画面は、ORで表示します。
- "月火水木金"などのグラフィックキャラクタはMSX BASICと異なり、直接10進数のキャラクタコードを代入できます。

# プログラム作成時のご注意

BASICコンパイラはMSX BASICと互換性があるように設計されていますが、多少の相違点があります。コンパイルしたときにエラーが発生する。コンパイルすると動作が異なる。このようなとき、次の点をチェックしてください。

## 変数関係

- BASICの部分とCALL TURBO ONの部分で変数、配列を共用する場合、あらかじめ定義された変数、配列のみが使用できます。もし、定義していない変数や配列を共用しようとするとき "Illegal function call" のエラーが起こります。変数や配列の定義が正しくされているか確認してください。
- 各命令の引数や添字の範囲のチェックは行われません（ただし配列の次元はコンパイル時にチェックされます）。MSX2、MSX2+ 特有の命令をコンパイルした場合の動作は保証されません。BASICコンパイラは動作の高速化のためにエラーチェックが簡略化されています。MSX BASICで正常に動作するのを確認してから、コンパイルしてください。
- FOR~NEXT文を正しく対応させてください。また、変数の省略はできません。
- MSX BASICでは、要素数が10までの配列は宣言しなくても使えますが、BASICコンパイラでは、配列は必ずDIM文を使って宣言しなければいけません。
- BASICコンパイラでの宣言(DEFINT、DEFSNG、DEFDBL、コメント中の#N)は、コンパイル中に効果が発生するのではなく、実行時に有効となります。このため、プログラム中に現れる順番に処理されます。例えば次のプログラムではMSX BASICとコンパイラでは実行結果が異なります。

```
10 GOSUB 50
20 A=100/3
30 PRINT A
40 END
50 DEFINT A
60 RETURN
```
- 文字列および文字型の変数、配列、関数、演算は普通のBASIC同様使えますが、一つの変数（または配列の一要素）ごとに256バイトを使用しますので、同じ変数名を繰り返し使うなど、メモリの使用に注意が必要です。



## 演算関係

- MSX BASICでは、10段の文字列のスタック(演算処理用のメモリ)を持っていて、これによって複雑な文字列演算を実行できますが、コンパイラでは文字列のスタックが1段しかありません。このため、MSX BASICで実行できた文字列の演算がコンパイルできないこともあります。  
\* 文字列のスタックを使う命令とは、文字列の連結(たし算)と文字列の比較(大小、長さ)です。

```
10 B$="41"  
20 A$=A$+C$+CHR$(VAL("&H"+B$))
```

上の例ではコンパイルを行うと、"String formula too complex in 20" というエラーが発生します。このエラーをさけるためには、文字列の演算をできるだけ分割するようにしてください。

上のプログラム例では次のように変更すると、正常に実行できます。

```
10 B$="41"  
20 C$=CHR$(VAL("&H"+B$))  
30 A$=A$+C$
```

- BASICコンパイラでは整数同士の演算結果は、割り算、べき乗の演算と特別の場合を除き、必ず整数になると仮定しています。  
例えばA%が200の時、A%\*A%はMSX BASICでは、浮動小数の40000になりますが、コンパイラでは-25536になります(オーバーフローを無視しているため)。  
演算の前に単精度実数の型宣言を入れ、1!\*A%\*A%とするとMSX BASICと同一の結果となります。  
「特別の場合」とは、「式の演算結果が浮動小数になるということが、前もって分かっている場合」です。  
例えば、上の例のように実数との演算、浮動小数変数への代入、浮動小数関数の引数の評価などを行うときです。ただし、この演算方法は完璧ではありません。LOG(1+A%\*A%)のように優先順位が高い演算が右にくる場合は、MSX BASICと同じ結果になりません。

## USR関数・プログラム制御

- USR関数の実行時のメモリマップはMSX BASICとBASICコンパイラで結果が違います。  
MSX BASICの時は、ページ1のメモリ上(&H4000~&H7FFF)にMSX BASICが存在しますが、コンパイラの時はコンパイラが存在します。
- USR関数に渡したり受け取ったりする値は、整数に限ります。
- 割り込みを使用するとプログラムサイズ、実行時間共に増加します。

## 速いプログラムを作るには

- 変数はできるだけ整数型にしてください。  
プログラムの初めに `DEFINT A-Z` 文などを設定してください。
- プログラム中の演算子には、「/」、「^」をなるべく使わず、「÷」、「\*」を使うようにします。
- 整数の割り算の時は、なるべく2のN乗の数(2, 4, 8, ...)で割るようにしてください。
- 整数の掛け算の時は、なるべく2のN乗の数か、3, 5, 6, 7, 9, 10, 20, 25, 40, 50, 80, 100, 200, 256, 257で掛けるようにしてください。
- 配列の要素数は、なるべく「上記の数-1」の値になるようにしてください。  
255がメモリの使用効率の点から、一番効率が良くなります。  
また多次元の配列を宣言する場合、この値の添字がなるべく左にくるように宣言してください。


## メモリの制限

- BASICコンパイラはコンパイルされる(BASICの)プログラムと、コンパイルされたプログラムが同じRAM(32Kバイト)上にあるので、メモリの使用できる制限がきびしくなります。  
大体の目安として10Kバイトぐらいのプログラムは、コンパイルして実行することができますが、次のような場合は動作しなくなることがあります。
  - ・ `CALL KANJI` 命令、ディスクドライブなどによってメモリのフリーエリアの上限が限られている場合。
  - ・ `CLEAR`、`MAXFILES` 命令などで余分にメモリが使われている場合。
  - ・ すでに非常に圧縮したBASICプログラムで、余分なスペースなどがほとんど入っていない場合。
  - ・ 割り込み処理を使ったプログラムの場合。
  - ・ 多くの文字列変数や、大きな配列を使っている場合。

以上のこと以外で `Out of memory` エラーとなるときは、できるだけ整数型の変数を使い、同じ処理をする部分はサブルーチン化し、余分なスペースはつめ、できる限りマルチステートメントを使って行数を減らすなどして、メモリを節約する必要があります。



# 資料



この章では、MSX-MUSIC、BASICコンパイラを使ったサンプルプログラムを記載しています。  
各機能を使ったプログラムの作成の参考にしてください。  
また、索引では、機能およびコマンドの説明してあるページを示しています。

サンプルプログラム .....	76
索引 .....	80

# サンプルプログラム

## MSX-MUSIC

FM音源のリズム音を使って、繰り返し演奏するプログラムです。

```
100 CALL MUSIC(1,0,1,1,1,1,1,1)
110 CALL BGM(1)
120 CALL VOICE(@48,@48,@6 ,@6 ,@14,@14)
130 T$="T150"
140 PLAY #2,T$,T$,T$,T$,T$,T$,T$
150 READ A$
160 IF A$="end" THEN RESTORE:GOTO 150
170 READ B$,C$,D$,E$,F$,G$
180 PLAY #2,A$,B$,C$,D$,E$,F$,G$
190 GOTO 150
200 DATA V15R1L8
210 DATA V15R1L8
220 DATA V15R1L8
230 DATA V15R1L8
240 DATA V12L1605D8DDD8DDD8DDD8DD
250 DATA V15L803DDDDDDDD
260 DATA V10CBH8H16H16SH8BH16H16BH8H16H1
6SH8BH16H16
270 DATA R805GR8GR8F4R8
280 DATA R805DR8DR8C4R8
290 DATA R804AR8AR8G4R8
300 DATA R1
310 DATA 05D80DD8DDD8DDD8DD
320 DATA D04C003D404C003D
330 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8BH16H16
340 DATA R1
350 DATA R1
360 DATA R1
370 DATA R1
380 DATA D8DDD8DDD8DDD8DD
390 DATA DDDDDDDDD
400 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8BH16H16
410 DATA R805GR8GR8F4R8
420 DATA R805DR8DR8C4R8
430 DATA R804AR8AR8G4R8
440 DATA R1
450 DATA 05D80DD8DDD8DDD8DD
460 DATA D04C003D404C003D
470 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8BH16H16
```



---

```
480 DATA R1
490 DATA R1
500 DATA R1
510 DATA R1
520 DATA D8DDD8DDD8DDD8DD
530 DATA DDDDDDDD
540 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8BH16H16
550 DATA R8O5GR8GR8F8R8
560 DATA R8O5DR8DR8C8R8
570 DATA R8O4AR8AR8G8R8
580 DATA R2R4R8
590 DATA O5D8DDD8DDD8DDD8
600 DATA D04CD03D404CD
610 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8
620 DATA D1R8
630 DATA O4A1R8
640 DATA F1R8
650 DATA R8R1
660 DATA DDD8DDD8DDD8DDD8DD
670 DATA O2B-4B-B-B-B-B-B-B-
680 DATA CBH4H16H16SH8BH16H16BH8H16H16SH
8BH16H16
690 DATA F1
700 DATA O5D1
710 DATA A1
720 DATA R1
730 DATA D8DDD8DDD8DDD8DD
740 DATA B-B-B-B-B-B-B-B-
750 DATA CBH8H16H16SH8BH16H16BH8H16H16SH
8BH16H16
760 DATA R1L205E-.C8D8
770 DATA R2R404B-105C8D8
780 DATA R2L204F.FC8D8
790 DATA R4L104C.C8D8
800 DATA O3G1R1
810 DATA O3C1R1
820 DATA R1R1
830 DATA end
```

## BASICコンパイラ

SCREEN 2で惑星(スプライト)を回転させます。

プログラムを実行すると、数字(表示のステップ)の入力を求めてきますので、100~300を指定してください。最大9個の惑星が回転します。

```
100 DEFINT A-Z:DEFSNG A,B,H,I,K,M,P
110 DIM C(14),X(300,10),Y(300,10)
120 COLOR 15,0,0:SCREEN 0
130 INPUT "NUMBER(100-300)";L
140 FOR J=20 TO L
150   IF L MOD J=0 THEN C(N)=J:N=N+1
160 NEXT J
170 PRINT N;"PLANETS"
180 C=N-1:H=90/N:N=0:D=L
190 '
200 P=ATN(1)*4
210 PRINT "WAIT"
220 FOR J=0 TO C
230   K=RND(1)*P/2:L=C(J)/2:M=H*(J+1)
240   FOR I=0 TO P*2 STEP P/L
250     PRINTC-J;P*2-I
260     A=COS(I+K):B=SIN(I)
270     X(N,J)=A*M*1.5+128
280     Y(N,J)=B*M+96
290     IF X(N,J)>255 OR X(N,J)<0 THEN X(N
,J)=-2:Y(N,J)=218
300     N=N+1
310   NEXT I
320   N=0
330 NEXT J
340 SCREEN 2,0
350 CIRCLE(128,96),4:PAINT(128,96)
360 SPRITE$(0)=CHR$(192)+CHR$(192)
370 IF INKEY$="" THEN 380 ELSE END
380 FOR J=0 TO C
390   PUT SPRITE J,(X(N MOD C(J),J),Y(N M
OD C(J),J)),15-J*2,0
400 NEXT J
410 N=(N+1) MOD D
420 GOTO 370
```



## 漢字モード

SCREEN 1と5を使ってバイオリズムを表示します。

プログラムを実行するまえに、SET DATE命令で今日の日付けに合わせてください。

プログラムを実行すると、あなたの誕生日を、きいてきますので入力してください。

```
100 '** バイオリズム クイズ **
110 CALL KANJI:CALL BC
120 SCREEN 1:WIDTH 60:DEFINT A-Z
130 GET DATE D$
140 TY=VAL("19"+LEFT$(D$,2))
150 TM=VAL(MID$(D$,4,2))
160 TD=VAL(RIGHT$(D$,2))
170 INPUT "あなたの誕生年は":YY
180 IF YY<1900 THEN YY=YY+1925
190 INPUT "          月は":YM
200 INPUT "          日は":YD
210 YT!=INT(365.25*YY)+INT(30.6*YM)+1+YD
220 TT!=INT(365.25*TY)+INT(30.6*TM)+1+TD
230 DS!=TT!-YT!
240 SCREEN 5:CALL KANJI1*
250 COLOR 2:LOCATE 4,1:PRINT "---身体"
260 COLOR 8:LOCATE 14,1:PRINT "---感情"
270 COLOR 11:LOCATE 24,1:PRINT "---知性"
280 LINE(0,120)-(255,120),15
290 LINE(40,40)-(40,180),15
300 COLOR 15:LOCATE 4,2:PRINT"今日"
310 FOR X=0 TO 255 STEP 4:IF(XMOD40)<>0 THEN L=14 ELSE L=15
320 T3=31:IF TM=2 THEN T3=28:IF (TYMOD4)=0 THEN T3=29
330 IF TM=4 OR TM=6 OR TM=9 OR TM=11 THEN T3=30
340 IF X<40 THEN T2=TD+X/4-10-1
350 IF(T2MODT3)=0THEN380
360 LINE(X,118)-(X,122),L
370 NEXT X:GOTO 420
380 TM=TM+1:TD=TD-T3:IF TM>12THEN TM=1
390 LINE(X,68)-(X,122),7:COLOR7
400 IF X>220 GOTO 370
410 LOCATE X/6,3:PRINT TM;"月":GOTO 370
420 R!=-2*3.1415926535898#
430 A=2:S=23:PSET(-5,120):GOSUB 500
440 A=8:S=28:PSET(-5,120):GOSUB 500
450 A=11:S=33:PSET(-5,120):GOSUB 500
460 LOCATE 2,4:PRINT"活発"
470 LOCATE 1,8:PRINT"不安定"
480 LOCATE 2,12:PRINT"低調";
490 A$=INKEY$:IF A$="" THEN 490 ELSE 540
500 FOR X=0 TO 260 STEP 4
510 P=45*SIN(R!*(DS!+X/4-10)/S)+120
520 LINE -(X,P),A
530 NEXT X:RETURN
540 END
```

# 索 引

## ア行

インタプリタ	60
音色データ	51, 54
音色パラメータ	54

## カ行

漢字の出力・制御	6, 26
漢字の種類	7, 26
漢字の入出力	6, 10
漢字の入力方法	6
漢字モードでの BASIC	11
画面の制御・関数	20
記号の入力	8
コンパイラ	59, 60

## タ行

単漢字変換	6, 7
-------	------

## ハ行

変換テーブルに 表示される文字数	8
---------------------	---

## ラ行

リズム音	45
------	----

## 英数字

19268色の表示	13
BASICコンパイラ	59
BASE	22
CALL AKCNV	22
CALL ANK	23
CALL AUDREG	43
CALL BGM	44
CALL CLS	23
CALL JIS	24
CALL KACNV	25
CALL KANJI	6, 26
CALL KEXT	28
CALL KINSTR	28
CALL KLEN	29
CALL KMID	29
CALL KNJ	30

CALL KTYPE	30
CALL MUSIC	45
CALL PALETTE	31
CALL PITCH	47
CALL PLAY	48
CALL SJIS	31
CALL STOPM	42
CALL TEMPER	49
CALL TRANSPOSE	50
CALL VOICE	51
CALL VOICE COPY	54
CLS	32
COLOR	32
F M音源	41
LOCATE	33
MML	57
MSX2+ の機能	4
MSX-MUSIC	41
PLAY	56
P S G音源	41
PUT KANJI	33
R G B方式	16
SCREEN	34
スクリーンの種類	15
SET SCROLL	36
VDP	37
WIDTH	38
Y J K方式	16
Y J K方式のデータ	16, 19





三洋電機株式会社